HP 4396A Network/Spectrum Analyzer

HP-IB Programming Guide

SERIAL NUMBERS

This manual applies directly to instruments with serial number prefix 3241J.
For additional important information about serial
numbers, read "Serial Number" in Appendix A.

HP Part No. 04396-90003
Microfiche Part No. 04396-90053
Printed in JAPAN December 1992

Notice

The information contained in this document is subject to change without notice.

Yokogawa-Hewlett-Packard, LTD.
Kobe Instrument Division
1-3-2, Murotani, Nishi-ku, Kobe-shi,
Hyogo, 651-22 Japan

Manual Printing History

The manual printing date and part number indicate its current edition. The printing date changes when a new edition is printed. (Minor corrections and updates that are incorporated at reprint do not cause the date to change.) The manual part number changes when extensive technical changes are incorporated.

December 1992..... 1st. Edition

Safety Summary

The following general safety precautions must be observed during all phases of operation, service, and repair of this instrument. Failure to comply with these precautions or with specific *WARNINGS* given elsewhere in this manual violates safety standards of design, manufacture, and intended use of the instrument.
The Hewlett-Packard Company assumes no liability for the customer's failure to comply with these requirements.

Ground The Instrument

This is a Safety Class 1 product (provided with a protective earth terminal). An uninterruptible safety earth ground must be provided from the main power source to the product input wiring terminals, power source to the product input wiring terminals, power cord, or supplied power cord set. Whenever it is likely that the protection has been impaired, the product must be made inoperative and secured against any unintended operation.

DO NOT Operate in an Explosive Atmosphere

Do not operate the instrument in the presence of flammable gases or fumes. Operation of any electrical instrument in such an environment is a safety hazard.

Keep Away From Live Circuits

Operating personnel must not remove instrument covers. Component replacement and internal adjustments must be made by qualified maintenance personnel. Do not replace components with the power cable connected. Under certain conditions, dangerous voltages may exist even with the power cable removed. To avoid injuries, always disconnect power and discharge circuits before touching them.

DO NOT Service Or Adjust Alone

Do not attempt internal service or adjustment unless another person, capable of rendering first aid and resuscitation, is present.

DO NOT Substitute Parts Or Modify Instrument

Because of the danger of introducing additional hazards, do not substitute parts or perform unauthorized modifications to the instrument. Return the instrument to a Hewlett-Packard Sales and Service Office for service and repair to ensure the safety features are maintained.

Dangerous Procedure Warnings

Warnings, such as the example below, precede potentially dangerous procedures throughout this manual. Instructions contained in the warnings must be followed.

Warning

Dangerous voltages, capable of causing death, are present in this instrument. Use extreme caution when handling, testing, and adjusting this instrument.

Typeface Conventions

Bold	Boldface type is used when a term is defined. For example: **icons** are symbols.
Italics	Italic type is used for emphasis and for titles of manuals and other publications. Italic type is also used for keyboard entries when a name or a variable must be typed in place of the words in italics. For example: `copy` *`filename`* means to type the word `copy`, to type a space, and then to type the name of a file such as `file1`.
`Computer`	Computer font is used for on-screen prompts and messages.
HARDKEYS	Labeled keys on the instrument front panel are enclosed in ▢.
SOFTKEYS	Softkeys located to the right of the CRT are enclosed in ▒.

Certification

Hewlett-Packard Company certifies that this product met its published specifications at the time of shipment from the factory. Hewlett-Packard further certifies that its calibration measurements are traceable to the United States National Institute of Standards and Technology, to the extent allowed by the Institution's calibration facility, or to the calibration facilities of other International Standards Organization members.

Warranty

This Hewlett-Packard instrument product is warranted against defects in material and workmanship for a period of one year from the date of shipment, except that in the case of certain components listed in *General Information* of this manual, the warranty shall be for the specified period. During the warranty period, Hewlett-Packard Company will, at its option, either repair or replace products which prove to be defective.

For warranty service or repair, this product must be returned to a service facility designated by HP. Buyer shall prepay shipping charges to HP and HP shall pay shipping charges to return the product to Buyer. However, Buyer shall pay all shipping charges, duties, and taxes for products returned to HP from another country.

HP warrants that software and firmware designated by HP for use with an instrument will execute its programming instruction when property installed on that instrument. HP does not warrant that the operation of the instrument, or software, or firmware will be uninterrupted or error free.

Limitation of Warranty

The foregoing warranty shall not apply to defects resulting from improper or inadequate maintenance by Buyer, Buyer-supplied software or interfacing, unauthorized modification or misuse, operation outside of the environmental specifications for the product, or improper site preparation or maintenance.

No other warranty is expressed or implied. HP specifically disclaims the implied warranties of merchantability and fitness for a particular purpose.

Exclusive Remedies

The remedies provided herein are buyer's sole and exclusive remedies. HP shall not be liable for any direct, indirect, special, incidental, or consequential damages, whether based on contract, tort, or any other legal theory.

Assistance

Product maintenance agreements and other customer assistance agreements are available for Hewlett-Packard products.

For any assistance, contact your nearest Hewlett-Packard Sales and Service Office. Addresses are provided at the back of this manual.

Safety Symbols

General definitions of safety symbols used on equipment or in manuals.

Instruction manual symbol: the product is marked with this symbol when it is necessary for the user to refer to the instruction manual in order to protect against damage to the instrument.

Indicates dangerous voltage (terminals fed from the interior by voltage exceeding 1000 volts must be so marked).



Protective conductor terminal. For protection against electrical shock in case of a fault. Used with wiring terminals to indicate the terminal which must be connected to ground before operating equipment.

Low-noise or noiseless, clean ground (earth) terminal. Used for a signal common, as well as providing protection against electrical shock in case of fault. A terminal marked with this symbol must be connected to ground in the manner described in the installation (Operation) manual, and before operating the equipment.



Frame or chassis terminal. A connection to the frame (chassis) of the equipment which normally includes all exposed metal structures.

Alternating current (power line).

Direct current (power line).

Alternating or direct current (power line).

Warning

Warning denotes a hazard. It calls attention to a procedure, practice, condition or the like, which, if not correctly performed or adhered to, could result in injury or death to personnel.

Caution

Caution denotes a hazard. It calls attention to a procedure, practice, condition or the like, which, if not correctly performed or adhered to, could result damage to or destruction of part or all of the product.

Note

Note denotes important information. It calls attention to a procedure, practice, condition or the like, which is essential to highlight.

Other Graphic Symbols

General definitions of other graphic symbols used in manuals.

COMPUTER denotes information for a programmer using an external computer as the system controller.

iBASIC denotes information for a programmer using an analyzer with HP Instrument BASIC as the system controller.

How to Use This Manual

This manual provides an introduction to writing BASIC programs for the HP 4396A Network/Spectrum Analyzer (analyzer). To reduce the time required for you to learn how to write programs for the analyzer, the examples shown in this guide are supplied on sample disks. You can perform each example sequentially or you can select the examples that apply to your immediate needs and learn those techniques. Use the table of contents and the index to quickly locate these examples.

Also, depending upon your experience in writing BASIC programs using HP-IB commands, you may want to do one of the following:

1. If you are an experienced programmer and have programmed HP-IB systems before, you can scan the examples in this guide to find out how the analyzer can be used in your system. If you have never programmed an instrument similar to the analyzer, you can start at the beginning and do the examples that apply to your application.
2. If you are an experienced programmer, but do not have any knowledge of HP-IB commands, review some examples to decide where you need help. See the *HP-IB Command Reference* for additional information on HP-IB commands.
3. If you are not an experienced programmer and you do not have any knowledge of HP-IB commands, see the *HP-IB Command Reference* for a list of the documentation that you will need to review before using this guide.

The analyzer can also use HP Instrument BASIC (if option 1C2 is installed). Documentation for HP Instrument BASIC and the other manuals available for the analyzer is listed in the Documentation Map on the next page.

Documentation Map

The following manuals are available for the analyzer.

User's Guide (HP Part Number 04396-90001)

The User's Guide walks you through system setup and initial power-on, shows you how to make basic measurements, explains commonly used features, and contains typical application measurement examples. After you receive your analyzer, begin with this manual.

The Task Reference (HP Part Number 04396-90000)

The Task Reference helps you to learn how to use the analyzer. This manual provides simple step-by-step instructions without concepts.

Function Reference (HP Part Number 04396-90002)

The Function Reference describes all functions accessed from the front panel keys and softkeys. It also provides information on options and accessories available, specifications, system performance, and conceptual information about the analyzer's features.

HP-IB Programming Guide (HP Part Number 04396-90003)

The HP-IB Programming Guide shows how to write and use BASIC programs to control the analyzer.

HP-IB Command Reference (HP Part Number 04396-90004)

The HP-IB Command Reference provides a summary of all available HP-IB commands. It also provides information on the status reporting structure and the trigger system (these features conform to the SCPI standard).

Using HP Instrument BASIC with the HP 4396A (Option 1C2 only) (HP Part Number 04396-90005)

The Using HP Instrument BASIC with the HP 4396A describes how HP Instrument BASIC works with the analyzer.

HP Instrument BASIC Users Handbook (Option 1C2 only), (HP Part Number E2083-90000)

The HP Instrument BASIC Users Handbook introduces you to the HP Instrument BASIC programming language, provide some helpful hints on getting the most use from it, and provide a general programming reference. It is divided into three books, *HP Instrument BASIC Programming Techniques* , *HP Instrument BASIC Interface Techniques* , and *HP instrument BASIC Language Reference.*

Performance Test Manual (HP Part Number 04396-90100)

The Performance Test Manual provides the tests required to verify that the analyzer conforms to its published specifications.

Service Manual (Option 0BW only), (HP Part Number 04396-90101)

The Service Manual explains how to adjust, troubleshoot, and repair the analyzer.

Microfiche Copies of the Manual

Use the microfiche part number on the title page to order a package of 10 × 15 centimeter (4 × 6 inch) microfilm transparencies of this manual, the *User's Guide* , the *Task Reference*, the *Function Reference*, the *HP-IB Programming Guide*, the *HP-IB Command Reference*, the *Using HP Instrument BASIC with the HP 4396A*, the *Performance Test Manual*, and the *Service Manual.*

PREFACE

Using a Sample Program Disk

Two sample program disks are furnished with the analyzer. These disks contain the sample programs listed in this manual.

Sample program disk for an external controller

(HP part number 04396-61002, Disk 1 of 2)

Use this disk for the sample programs used with the external controller.



Sample program disk for HP Instrument BASIC

(HP part number 04396-61002, Disk 2 of 2)

Use this disk for the sample programs used with HP Instrument BASIC.

Note

If you are going to use HP Instrument BASIC, you must first allocate the analyzer's display format. Use `DISPLAY ALLOCATION` (under Display) and `MORE`) to set the display to `HALF INSTR HALF BASIC` or `ALL BASIC`. See the *Using HP Instrument BASIC with the HP 4396A* for additional information.

To Check the Files List

These programs are saved in ASCII format. To check the files list:

1. Put the program disk into the disk drive and type as follows.

```
CAT
```

2. Press Return.

```
CAT

FILE NAME  PRO TYPE REC/FILE BYTE/REC   ADDRESS    DATE    TIME

FIG1_3        ASCII       6      256        34   22-Jun-92 11:00
FIG2_2        ASCII       6      256        34   22-Jun-92 11:00
FIG2_3        ASCII       6      256        34   22-Jun-92 11:00
      :
```

Each file name also represents the figure number in which the program list is shown in this guide. For example, the sample program listed in Figure 4-3 is saved with the name `FIG4_3`.

To Get a Program

To get the program use the GET command. For example, to get the sample program FIG4_3:

1. Type as follows:

```
GET "FIG4_3"
```

2. Press Return.

Contents

Figures

Tables

1

Learning HP-IB Remote Control Basics

This chapter provides information on how to configure the HP-IB remote-control system and the basic use of the HP-IB commands. In the examples used in this manual, most of the commands are the simple HP-IB commands. For each of these commands, there is also a corresponding command that conforms to the Standard Commands for Programmable Instruments (SCPI) standard. For additional information of about all commands, see the *HP-IB Command Reference* manual.

What is HP-IB?

The Hewlett-Packard Interface Bus (HP-IB) is used for remote control of the HP 4396A Network/Spectrum Analyzer (analyzer). HP-IB is a standard for interfacing instruments to computers and peripherals. This standard supports worldwide standards IEEE 488.1, IEC-625, IEEE 488.2, and JIS-C1901. The HP-IB interface allows the analyzer to be controlled by an external computer. The computer sends commands or instructions to and receives data from the instrument through the HP-IB.

Required Equipment

To perform the examples in this manual, you need the following equipment:

1. The analyzer and the accessories required to test a specific device under test (DUT).
2. For the HP-IB system controller,



If the analyzer has the HP Instrument BASIC (Option 1C2) installed, it can be used as the system controller.

Or,

An HP 9000 Series 200 or 300 computer or an HP Vectra PC with a measurement coprocessor or card (HP 82300 or 82324). The computer must have enough memory to hold BASIC, needed binaries, and at least 64 kilobytes of program space.

BASIC 3.0 or higher operating system and the following binary extensions:

HPIB, GRAPH, IO, KBD, and ERR

A disk drive is required to load BASIC, if no internal disk drive is available. (Depending on the disk drive, a binary such as CS80 may be required.)

3. Peripherals (printer, plotter, and so on) and any HP-IB instruments that are required for your application.
4. HP 10833A/B/C/D HP-IB cables to interconnect the computer, the analyzer, and any peripherals.

To Prepare for HP-IB Control

1. Connect the analyzer and controller, plus any other instruments and peripherals with HP-IB cables.



Figure 1-1. System Configuration for HP-IB Remote Control

* To set printer or plotter see Chapter 6.

2. Turn on the analyzer.
3. Prepare the system controller.

iBASIC

If you are using only HP Instrument BASIC and no external controller, prepare the analyzer for your use. For details, see *Using HP Instrument BASIC with the HP 4396A*.

COMPUTER

If you are using a computer as an external controller,

a. Set the analyzer to addressable only mode.

Press (Local) ADDRESSABLE ONLY.

b. Set HP-IB address of the analyzer to 17.

Press (Local) SET ADDRESS ADDRESS: 4396 (1) (7) (x1).

c. Turn on the controller. Then load the BASIC operating system and the binary extensions.

How large a system can you configure?

- A maximum of 15 devices can be connected on one bus system.
- The length of cable between one device and another must be less than or equal to four meters. The total length of cable in one bus system must be less than or equal to two meters times the number of devices connected on the bus (the HP-IB controller counts as one device). The total length of cable must not exceed 20 meters.
- Star, linear, and combinational cable configurations are allowed. There must be no loop.



- It is recommended that no more than four piggyback connectors be stacked together on one device. Otherwise, the resulting structure could exert enough force on the connector mounting to damage it.

HP-IB Commands Introduction

All the analyzer's front-panel keys have a corresponding HP-IB command. By executing an HP-IB command, you can operate the analyzer as if you were pressing the corresponding key.

For example,

Pressing (Preset) is the same as executing the HP-IB command, `PRES`.

Note

Each of the analyzer's functions has two corresponding HP-IB commands: One is unique to analyzer, and the other corresponds to the SCPI (Standard Commands for Programmable Instruments) standard. In this guide, only the commands that are unique to the analyzer are described.

For example, you can use either `NA` (which is unique to the analyzer) or `INST:TYPE NA` (which conforms to SCPI) to select the analyzer type.

For details on SCPI, see *HP-IB Command Reference.*

To Execute an HP-IB Command

Combine the BASIC `OUTPUT` statement with the HP-IB select code, the device address, and finally the analyzer command. For example, to execute `PRES` command, type:

Using HP Instrument BASIC



And press (Return). The analyzer goes to the preset state.

Using an External Controller



And press (Return). The analyzer is set to HP-IB remote mode. Then the analyzer goes to the preset state.

What is HP-IB remote mode?



Executing an `OUTPUT` statement that is addressed to the analyzer, sets it to the HP-IB remote mode. In the remote mode, all the analyzer's front-panel keys are locked out, except (Local). Pressing (Local) puts the analyzer back in local mode. In local mode, all front-panel keys are enabled.

To Program a Basic Measurement

This section describes how to organize the commands into a measurement sequence. Figure 1-2 shows a typical program flow for a measurement.



Figure 1-2. Program Flow

The following program performs the measurement flow controlling the analyzer using HP-IB.

This guide shows program lists of sample programs for an external controller. To use the sample programs in this guide with HP Instrument BASIC, change the select code from 7 to 8 and change the HP-IB address from 17 to 00 (that is, use 800 instead of 717).

```
10    !
20    ! Figure 1-3. Basic Measurement
30    !
40    ASSIGN @Hp4396 TO 717 ! When iBASIC is used, change "717" to "800".
50    !
60    OUTPUT @Hp4396;"PRES" ! Preset HP 4396A
70    OUTPUT @Hp4396;"CHAN1;NA;MEAS S21;FMT LOGM"
80    INPUT "Enter center frequency (Hz).",F_cent
90    INPUT "Enter frequency span (Hz).",F_span
100   OUTPUT @Hp4396;"CENT ";F_cent
110   OUTPUT @Hp4396;"SPAN ";F_span
120   !
130   ! Frequency Response Calibration
140   OUTPUT @Hp4396;"CALK N50"       ! Select 50 ohm type-N Cal. kit
150   OUTPUT @Hp4396;"CALI RESP"      ! Select Response cal.
160   OUTPUT @Hp4396;"CLES"           ! Clear all status
170   INPUT "Connect THRU, then press [Return].",Dum$
180   OUTPUT @Hp4396;"*SRE 4;ESNB 1" ! Set enable STB and ESB
190   ON INTR 7 GOTO Cal_end     ! \ When iBASIC is used, change "7" to "8".
200   ENABLE INTR 7;2            ! /
210     OUTPUT @Hp4396;"STANC"         ! Measure THRU
```

Figure 1-3. Sample Program : Basic Measurement (1/2)

```
220 Calibrating:    GOTO Calibrating
230 Cal_end:    !
240   OUTPUT @Hp4396;"RESPDONE"      ! Calculating cal coefficients
250   OUTPUT @Hp4396;"*OPC?"       ! \ Waiting calculation end
260   ENTER @Hp4396;Dum            ! /
270   DISP "Response cal completed."
280   !
290   ! Measurement
300   INPUT "Connect DUT, then press [Return].",Dum$
310   OUTPUT @Hp4396;"CLES"    ! Clear all status registers
320   OUTPUT @Hp4396;"*SRE 4;ESNB 1"
330   ON INTR 7 GOTO Sweep_end     ! \ When iBASIC is used,
340   ENABLE INTR 7;2              ! / change "7" to "8"
350     OUTPUT @Hp4396;"SING"    ! Sweep mode is SINGLE
360 Measuring:    GOTO Measuring
370 Sweep_end:   !
380   OUTPUT @Hp4396;"MKR ON"        ! Marker 1 ON
390   OUTPUT @Hp4396;"SEAM MAX"      ! Search MAX
400   OUTPUT @Hp4396;"OUTPMKR?"      ! Output marker value
410   ENTER @Hp4396;Val1,Val2,Swp
420   PRINT "Max val:",Val1;"dB"
430   PRINT "Swp.Prmtr:",Swp;"Hz"
440   END
```

Figure 1-3. Sample Program : Basic Measurement (2/2)

Set I/O Path

```
40    ASSIGN @Hp4396 TO 717 ! When iBASIC is used, change "717" to "800".
```

This operation allows you to use @Hp4396, instead of 717(or 800), as the HP-IB address in the program.

Set Up the Measurement Parameters

```
60    OUTPUT @Hp4396;"PRES" ! Preset HP 4396A
70    OUTPUT @Hp4396;"CHAN1;NA;MEAS S21;FMT LOGM"
80    INPUT "Enter center frequency (Hz).",F_cent
90    INPUT "Enter frequency span (Hz).",F_span
100   OUTPUT @Hp4396;"CENT ";F_cent
110   OUTPUT @Hp4396;"SPAN ";F_span
```

You can execute HP-IB commands in the same sequence as key operation. Lines 60 and 70 perform the same operation as pressing (Preset) (Ch1) (Meas) ANALYZER TYPE NETWORK ANALYZER S PARAMETERS Trans:FDW S21[B/R] (Format) LOG MAG.

In general, the procedure for setting up measurements on the analyzer via HP-IB follows the same sequence as performing the procedure manually. There is no required order, as long as the desired frequency range, number of points, and power level are set before performing the calibration.

In line 70, several HP-IB commands, separated by semicolon, are executed in a line. This is the same as:

```
70    OUTPUT @Hp4396;"CHAN1"
71    OUTPUT @Hp4396;"NA"
72    OUTPUT @Hp4396;"MEAS S21"
73    OUTPUT @Hp4396;"FMT LOGM"
```

In lines 80 to 110 (setting frequency), parameters are required with the HP-IB command. To set parameters, see "To Execute an HP-IB Command with a Parameter" later in this chapter.

Perform Calibration

```
130   ! Frequency Response Calibration
140   OUTPUT @Hp4396;"CALK N50"       ! Select 50 ohm type-N Cal. kit
150   OUTPUT @Hp4396;"CALI RESP"      ! Select Response cal.
160   OUTPUT @Hp4396;"CLES"           ! Clear all status
170   INPUT "Connect THRU, then press [Return].",Dum$
180   OUTPUT @Hp4396;"*SRE 4;ESNB 1" ! Set enable STB and ESB
190   ON INTR 7 GOTO Cal_end     ! \ When iBASIC is used, change "7" to "8".
200   ENABLE INTR 7;2            ! /
210     OUTPUT @Hp4396;"STANC"         ! Measure THRU
220 Calibrating:    GOTO Calibrating
230 Cal_end:    !
240   OUTPUT @Hp4396;"RESPDONE"      ! Calculating cal coefficients
250   OUTPUT @Hp4396;"*OPC?"      ! \ Waiting calculation end
260   ENTER @Hp4396;Dum           ! /
270   DISP "Response cal completed."
```

In lines 140 to 240, the HP-IB program follows the key strokes required to calibrate from the front panel. This program performs a response calibration.

Line 170 requests the operator to connect a THRU calibration standard.

Lines 180 through 220 use the status bytes to detect the completion of the THRU calibration. See "To Wait for Sweep End" in Chapter 3.

Lines 240 through 270 use the *OPC? command to detect the completion of the calculation of the calibration coefficients. See "To Wait For the Preceding Operation to Complete" in Chapter 3.

Connect DUT

```
300   INPUT "Connect DUT, then press [Return].",Dum$
```

Line 300 requests the operator to connect a DUT to the analyzer.

All instrument settings and calibration are done. You can now measure the DUT.

Trigger a Measurement

```
310   OUTPUT @Hp4396;"CLES"   ! Clear all status registers
320   OUTPUT @Hp4396;"*SRE 4;ESNB 1"
330   ON INTR 7 GOTO Sweep_end    ! \ When iBASIC is used,
340   ENABLE INTR 7;2             ! / change "7" to "8"
350     OUTPUT @Hp4396;"SING"   ! Sweep mode is SINGLE
360 Measuring:    GOTO Measuring
370 Sweep_end:   !
```

Lines 310 to 370 enable SRQ interruption for sweep end detection. For details, see Chapter 3.

In line 350, the analyzer executed a single trigger. For more advanced trigger control, see Chapter 2.

Post-Processing

```
380   OUTPUT @Hp4396;"MKR ON"       ! Marker 1 ON
390   OUTPUT @Hp4396;"SEAM MAX"      ! Search MAX
```

Line 380 activates the marker and line 390 moves the marker to the maximum value on the trace. For details on using the marker, see Chapter 4.

Transfer Data

```
400   OUTPUT @Hp4396;"OUTPMKR?"  ! Output marker value
410   ENTER @Hp4396;Val1,Val2,Swp
```

Line 400 the measured data is transferred to the controller. For details about data transfer, see Chapter 4.

To Execute an HP-IB Command with a Parameter

Some HP-IB commands require a numeric parameter. For example:

```
OUTPUT @Hp4396;"CENT 25000000"   ! Set center frequency to 25 MHz.
```

(The space between the command and the numeric parameter is mandatory.)

You can program it to be entered each time the program is run. For example:

```
100 INPUT "Enter center frequency(Hz).";F_cent
110 OUTPUT @Hp4396;"CENT ";F_cent
```

Executing this,

```
Enter center frequency (Hz).
25000000
```

The analyzer's center frequency is set to 25 MHz.

To Execute a Query

Any HP-IB command that is used with a numeric parameter can also be used as query command. For example, the CENT *numeric_parameter* command used in the previous example, can be combined with a ?, and used as a query command as follows,

```
10 OUTPUT @Hp4396;"CENT?"
20 ENTER @Hp4396;A
30 PRINT A
```

The CENT? command returns the current center frequency, which is put into A. Executing this program results in the following:

```
25000000
```

By interrogating the analyzer to determine the values of the start and stop frequencies, or the center frequency and frequency span, the computer can keep track of the actual frequencies.

2

Triggering the Analyzer from Remote

This chapter describes how to control the trigger system of the analyzer.

To trigger a measurement from a controller, the following steps are commonly used:

1. Set the trigger source to:

   Bus, or Internal (free run)

   (In External, Video, Manual or Gate trigger, you cannot trigger from the controller, so these sources are not mentioned in this guide.)

2. Set the number of measurements and the analyzer is initiated. You can set the number of measurements as:

   (Hold)—Single—Number of Group—Continuous

3. Generate the trigger event and the analyzer starts a measurement.

The analyzer trigger system has three states: Idle, Waiting for Trigger, and Measurement.



Figure 2-1. Trigger System

In Figure 2-1,

1. After a HOLD HP-IB command execution, the analyzer returns to the "Idle" state.
2. By setting the number of measurements, the analyzer changes from the "Idle" state to the "Waiting for Trigger" state.
3. At the "Waiting for Trigger" state, a trigger input (corresponding to the trigger source) starts a measurement.

Bus	HP-IB command *TRG or BASIC command TRIGGER triggers measurements.
Internal (free run)	There is no need for a trigger input. The analyzer starts the measurements immediately.

4. After the measurement is complete, the next state depends on the number of measurements.

Single	goes to the "Idle" state(4-a).
Number of Groups	Goes to the "Waiting for Trigger" state until the number of groups not measured yet equals zero(4-b). After all measurements are completed, goes to "Idle" state(4-a).
Continuous	goes to the "Waiting for Trigger" state(4-b).

To Measure Continuously

```
10 !
20 ! Figure 2-2. To Trigger Measurement Continuously
30 !
40 ASSIGN @Hp4396 TO 717  ! When iBASIC is used, change "717" to "800"
50 OUTPUT @Hp4396;"TRGS INT"
60 OUTPUT @Hp4396;"CONT"
70 END
```

Figure 2-2. Sample Program : To Trigger Measurements Continuously

Set Trigger Source

```
50 OUTPUT @Hp4396;"TRGS INT"
```

Set the trigger source to internal.

Start Continuous Measurement Sweep

```
60 OUTPUT @Hp4396;"CONT"
```

The analyzer changes to the "Waiting for Trigger" state. In this program, the internal trigger source is selected and the analyzer immediately starts continuous measurements.

What can you do to abort a measurement?

Send the command:

```
OUTPUT @Hp4396;"HOLD"
```

The measurement sweep is aborted.

What are other trigger commands?

Instead of CONT, you can use,

```
OUTPUT @Hp4396;"SING"                    for single measurement
OUTPUT @Hp4396;"NUMG parameter"          for number of group measurements
```

When you transfer measurement data to the controller, you must use either the SING or the NUMG *parameter* command to synchronize the controller and the analyzer. To use these commands, see the "To Trigger a Measurement From the Controller" example.

To Trigger a Measurement From the Controller

Two methods of triggering a measurement from the controller are shown in Figure 2-3 and Figure 2-4.

```
10 !
20 ! Figure 2-3. To Trigger Measurement From Controller(1)
30 !
40 ASSIGN @Hp4396 TO 717 ! When iBASIC is used, change "717" to "800"
50 OUTPUT @Hp4396;"TRGS INT"
60 OUTPUT @Hp4396;"SING"
70 END
```

Figure 2-3. Sample Program : To Trigger a Measurement from Controller (1)

Set Trigger Source

```
50 OUTPUT @Hp4396;"TRGS INT"
```

Set the trigger source to internal.

Trigger a Measurement

```
60 OUTPUT @Hp4396;"SING"
```

The analyzer changes to the "Waiting for Trigger" state. In this program, the internal source is selected and the analyzer immediately starts a measurement. After the measurement, the analyzer goes to the "Idle" state.

How can you perform averaging?

When you set the averaging on, you must also set the number of measurements to the same value as the averaging factor. For example, if the averaging factor is 10, replace line 60 as follows:

```
60 OUTPUT @Hp4396;"NUMG 10"
```

How can you wait for a measurement to be completed?

When you want to return the measurement data to the controller, you must wait for the measurement to be completed. For details, see Chapter 3.

```
10 !
20 ! Figure 2-4. To Trigger Measurement From the Controller(2)
30 !
40 ASSIGN @Hp4396 TO 717 ! When iBASIC is used, change "717" to "800"
50 OUTPUT @Hp4396;"TRGS BUS"
60 OUTPUT @Hp4396;"CONT"
70 OUTPUT @Hp4396;"*TRG"
80 END
```

Figure 2-4. Sample Program : To Trigger a Measurement from Controller (2)

Set Trigger Source

```
50 OUTPUT @Hp4396;"TRGS BUS"
```

Set the trigger source to bus.

Trigger a Measurement

```
70 OUTPUT @Hp4396;"*TRG"
```

Triggers the analyzer. When the trigger source is set to bus, you can use the group execution trigger as follows:

```
70 TRIGGER 7
```

What is Group Execution Trigger (GET)?



The HP BASIC command TRIGGER can be used instead of the *TRG command. The HP BASIC command is used to trigger all triggerable instruments on a BUS at the same time. Therefore, to trigger all triggerable instruments on select code 7(HP-IB bus) execute the command:

```
TRIGGER 7
```

3

Synchronizing the Analyzer from Remote

HP-IB analyzer control programs that can be used for guiding you through an analyzer calibration, and for reading and manipulating measurement data.

The control program must wait until the calculation data is processed before continuing with the next instruction. Also, the control program must wait until the measurement is completed before it reads the measurement data.

This chapter describes two techniques used by the control programs to synchronize the controller and the analyzer.

- Uses the `*OPC?` command.

  This command halts the execution of the program until the analyzer completes the preceding commands in the program.

- Reports the analyzer's status and generates SRQ.

  The analyzer has a status reporting mechanism that gives information about specific functions and events inside the analyzer. The status byte is an 8-bit register with each bit summarizing the state of one aspect of the analyzer. For example, the error queue summary bit is set if there are any errors in the queue. For the status byte register bit assignment, see the *HP-IB Command Reference* manual.

What is an SRQ?

An SRQ (Service Request) is an interrupt generated by the analyzer. The analyzer can be setup to sent an SRQ when it needs the attention of the controller. The controller can ignore the SRQ or it can be setup to interrupt the program using the ON INTR commands. The Status Byte can be used to define the specific event that generates an SRQ (for example, the end of sweep complete).



Figure 3-1. SRQ Generation

To Wait For the Preceding Operation to Complete

```
10 !
20 ! Figure 3-2. To Wait for the Preceding Operation Complete
30 !
40   ASSIGN @Hp4396 TO 717     ! When iBASIC is used, change "717" to "800"
50   !
60   !  OUTPUT statement to send HP-IB command
70   !
80   OUTPUT @Hp4396;"*OPC?"
90   ENTER @Hp4396;A
100  !
110  !  Next operation
120  !
130  END
```

Figure 3-2. Sample Program: To Wait for the Preceding Operation to Complete

Let Controller Wait For Operation to Complete (OPC)

```
80   OUTPUT @Hp4396;"*OPC?"
90   ENTER @Hp4396;A
```

In line 80, the *OPC? command waits for the preceding operations to complete and then returns a 1.

In line 90, the controller pauses the program until the analyzer returns a 1.

For example, in the sample program in Figure 1-3 (Chapter 1), the *OPC? command is used as follows:

```
:
240   OUTPUT @Hp4396;"RESPDONE"      ! Calculating cal coefficients
250   OUTPUT @Hp4396;"*OPC?"       ! \ Waiting calculation end
260   ENTER @Hp4396;Dum            ! /
270   DISP "Response cal completed."
:
```

You cannot use *OPC? for the functions listed under SRQ (at the beginning of the chapter). Use the status byte for these functions.

To Wait for Sweep End

```
10  !
20  ! Figure 3-3. To Wait for Sweep End
30  !
40   ASSIGN @Hp4396 TO 717  ! When iBASIC is used, change "717" to "800"
50   OUTPUT @Hp4396;"TRGS INT"
60   OUTPUT @Hp4396;"CLES"
70   OUTPUT @Hp4396;"*SRE 4;ESNB 1"
80   ON INTR 7 GOTO Sweep_end  ! \ When iBASIC is used, change "7" to "8"
90   ENABLE INTR 7;2           ! /
100    OUTPUT @Hp4396;"SING"
110 Measuring:GOTO Measuring
120 Sweep_end:  !
130  DISP "MEASUREMENT COMPLETE"
140  END
```

Figure 3-3. Sample Program : To Wait for Sweep End

Enable Sweep-End Bit

```
60   OUTPUT @Hp4396;"CLES"
70   OUTPUT @Hp4396;"*SRE 4;ESNB 1"
```

Line 60 clears all bits of the Status Registers and the Enable Registers.

In line 70, the command *SRE 4 sets the Service Request Enable Register to 00000100 (this enables bit 2 of the Status Byte Register). The command ESNB 1 sets the Event Status Enable Register B to 0000000000000001 (this enables bit 0 of the Event Status Register B. See Figure 3-4).



Figure 3-4. Sweep-End Bit Enabling

Enable Registers select which events in the analyzer can cause a service request (SRQ). By setting bit 0 of the Event Status Enable Register B to 1, the occurrence of the corresponding event (sweep-end) sets bit 0 of the Event Status Register B. When this bit is set (and is enabled), it is used to set a summary bit in the Status Byte Register (bit 2). Also, because bit 2 of Service Request Enable Register is set, setting the corresponding bit (Event Status Register B summary bit) generates an SRQ. The SRQ sets bit 6 of the Status Byte Register.

Enable SRQ Interrupt

```
80   ON INTR 7 GOTO Sweep_end  ! \ When iBASIC is used, change "7" to "8"
90   ENABLE INTR 7;2           ! /
:
120 Sweep_end:  !
```

Line 80 defines a branch. When the SRQ interrupt is generated from the HP-IB interface (whose select code is 7), the controller goes to `Sweep_end` (Line 120).

Line 90 enables an interrupt from interface 7 (HP-IB) when bit 1 (SRQ bit) of the interrupt register (of the controller) is set by a value of 2. See the *HP-IB Command Reference* for additional information.

Wait Until Measurement Is Done

```
100    OUTPUT @Hp4396;"SING"
110 Measuring:     GOTO Measuring
```

In line 100, the `SING` command triggers a measurement and the analyzer starts a sweep. For details on how to trigger a measurement, see Figure 2-3.

The controller loops back in line 110 until an SRQ interrupt occurs.

Generate SRQ

On a single sweep end, bit 0 of the ESB is set (which sets bit 2 of the Status Byte Register) and an SRQ is generated.

```
110 Measuring:     GOTO Measuring   Loop until SRQ interrupt
120 Sweep_end:  !                   At SRQ interrupt, jump to here
```

Once an SRQ is generated, the SRQ interrupt is disabled.

To Report Command Error Occurrence

```
10   !
20   ! Figure 3-5. To Report Command Error Occurrence
30   !
40   ASSIGN @Hp4396 TO 717  ! When iBASIC is used, change "717" to "800"
50   !
60   DIM Err$[30]
70   OUTPUT @Hp4396;"CLES"
80   OUTPUT @Hp4396;"*SRE 32 ;*ESE 32"
90   ON INTR 7 GOSUB Err_report   ! \ When iBASIC is used,
100  ENABLE INTR 7;2              ! / change "7" to "8"
110     !
120     !  OUTPUT statement to send HP-IB command
130     !
140     !
150     !
160   GOTO Prog_end
170 Err_report:    !
180     OUTPUT @Hp4396;"OUTPERRO?"
190     ENTER @Hp4396;Err,Err$
200     PRINT "COMMAND ERROR DETECTED"
210     PRINT Err,Err$
220     !
230     A=SPOLL(@Hp4396)
240     OUTPUT @Hp4396;"*ESR?"
250     ENTER @Hp4396;Estat
260     ENABLE INTR 7    ! When iBASIC is used, change "7" to "8"
270   RETURN
280 Prog_end:   !
290   END
```

Figure 3-5. Sample Program : To Report Command Error Occurrence

For details on SRQ interrupt, see the "To Wait for Sweep End" example.

Enable Error Bit

```
70   OUTPUT @Hp4396;"CLES"
80   OUTPUT @Hp4396;"*SRE 32 ;*ESE 32"
```

Line 70 clears all bits of the Status Registers and Enable Registers.

In line 80, the command *SRE 32 sets the Service Request Enable Register to 00100000 (this enables bit 5 of the Status Byte Register). The command *ESE 32 sets the Standard Event Status Enable Register to 00100000 (this enables bit 5 of the Standard Event Status Register (see Figure 3-6).



Figure 3-6. Command-Error Bit Enabling

Report Command Error

```
90   ON INTR 7 GOSUB Err_report    ! \ When iBASIC is used,
100  ENABLE INTR 7;2               ! / change "7" to "8"
110     !
120     !  OUTPUT statement to send HP-IB command
130     !
140     !
150     !
160   GOTO Prog_end
170 Err_report:     !
```

If an HP-IB command (executed between lines 100 and 160) causes an HP-IB command error, the analyzer generates an SRQ and the controller branches to `Err_report`. For example, the `OUTPUT` statement:

```
120 OUTPUT @Hp4396;"CENT "  Setting center frequency, but no parameter
```

causes an SRQ interrupt and branch to `Err_report`.

Output Error

```
180     OUTPUT @Hp4396;"OUTPERRO?"
190     ENTER @Hp4396;Err,Err$
200     PRINT "COMMAND ERROR DETECTED"
210     PRINT Err,Err$
```

These commands retrieve the error number and description.

In the error shown in the line 120 example, the controller displays the following:

```
COMMAND ERROR DETECTED
-109   "Missing parameter"
```

Return to Execute HP-IB command

```
230      A=SPOLL(@Hp4396)
240      OUTPUT @Hp4396;"*ESR?"
250      ENTER @Hp4396;Estat
260      ENABLE INTR 7    ! When iBASIC is used, change "7" to "8"
270    RETURN
```

Lines 230 to 270 clear SRQ before returning to the main routine.

Line 230 reads the analyzer's status byte. The `A=SPOLL(@Hp4396)` statement reads the Status Byte Register of the address `@Hp4396`(analyzer), and enters the value into `A`. The command error causes the SRQ and with bit 5 and bit 6 of the Status Byte Register set, the value of `A` is 96. Reading the Status Byte Register by using the `SPOLL` command clears SRQ (status byte bit 6).

In line 240 and line 250, the command `*ESR?` reads the contents of the Standard Event Status Register. With Bit 5 of Standard Event Status Register set, the value of `Estat` is 32. Reading the Standard Event Status Register by using the `*ESR?` command clears the register.

A branch to `Err_report` disables the interrupt. Therefore, the return from `Err_report` must reenable the interrupt.

4

Reading Measurement Data

This chapter describes how to read measurement data over the HP-IB.

Measurement data can be read out of the analyzer in the following ways:

1. Data can be read off the trace selectively using the markers.

   The present value of the marker (real-imaginary data and sweep parameter) is retrieved. For additional information on the marker functions, see the *Function Reference*.

2. The entire trace (or data for a specified number of points) can be read out in the following ways:

   - Data arrays — In regard to the data processing flow, the following data arrays are available.

     RAW DATA ARRAYS
     CALIBRATION COEFFICIENT ARRAYS
     DATA ARRAYS
     MEMORY ARRAYS
     DATA TRACE ARRAYS
     MEMORY TRACE ARRAYS

     For details about the data processing flow of the analyzer, see Chapter 12 of *Function Reference* manual.

   - Data format — The analyzer provides four data transfer formats.

     FORM2 IEEE 32 bit floating point format
     FORM3 IEEE 64 bit floating point format
     FORM4 ASCII format
     FORM5 MS-DOS® personal computer format

     Depending on the format, the data transfer speed and the number of digits are changed. Generally, binary data transfer (FORM2, FORM3, or FORM5) is faster than ASCII (FORM4).

     For details on data transfer format, see the *HP-IB Command Reference* manual.

To Read Data Using the Marker Search Function

```
10    !
20    ! Figure 4-1. To Read Data Using Marker Search Function
30    !
40    ASSIGN @Hp4396 TO 717    ! When iBASIC is used, change "717" to "800"
50    INPUT "ENTER CENTER FREQUENCY (Hz)",F_cent    !  Setting HP 4396A
60    INPUT "ENTER FREQUENCY SPAN (Hz)",F_span      !
70    OUTPUT @Hp4396;"CENT ";F_cent                 !
80    OUTPUT @Hp4396;"SPAN ";F_span                 !
90    OUTPUT @Hp4396;"*OPC?"
100   ENTER @Hp4396;Dum
110  !
120   OUTPUT @Hp4396;"CLES"
130   OUTPUT @Hp4396;"*SRE 4;ESNB 1"
140   ON INTR 7 GOTO Sweep_end    ! \ When iBASIC is used, change "7" to "8"
150   ENABLE INTR 7;2             ! /
160     OUTPUT @Hp4396;"SING"         ! Trigger a Measurement
170 Measuring:    GOTO Measuring      ! Measuring
180 Sweep_end:   !
190     OUTPUT @Hp4396;"MKR ON"
200     OUTPUT @Hp4396;"SEAM MAX"
210     OUTPUT @Hp4396;"OUTPMKR?"
220     ENTER @Hp4396;Val1,Val2,Swp
230     PRINT "Max Val:",Val1;"dB"
240     PRINT "Swp.Prmtr:",Swp,"Hz"
250   END
```

Figure 4-1. Sample Program : To Read Data Using Marker Search Function

Search Maximum Value

```
190     OUTPUT @Hp4396;"MKR ON"
200     OUTPUT @Hp4396;"SEAM MAX"
```

Line 190 activates the marker and line 200 moves the marker to the maximum value on the trace.



Marker on Trace

What are the other marker commands?

You can activate sub-markers and the Δmarker using the following commands:

`SMKR{1-7} ON, DMKR {ON|FIX|TRAC}`

You can move the marker using the following commands:[1]

- specified sweep parameter — `MKRPRM` *parameter*
- specified measurement point — `MKRP` *parameter*

You can move sub-markers using the following commands:[1]

- specified sweep parameter — `SMKRPRM{1-7}` *parameter*
- specified measurement point — `SMKRP{1-7}` *parameter*

You can move the Δmarker using the following commands:[1]

- specified sweep parameter — `DMKRPRM` *parameter*
- specified primary part of marker value — `DMKRVAL` *parameter*
- specified secondary part of marker value — `DMKRAUV` *parameter*

1 Before executing these commands, you must turn on the markers to be moved.

Read Data

```
210     OUTPUT @Hp4396;"OUTPMKR?"
220     ENTER @Hp4396;Val1,Val2,Swp
```

The `OUTPMKR?` command returns the marker value in the following order: primary part of data, secondary part of data, and sweep parameter.

What are other marker value commands?

You can get the marker value using the following commands:

- get primary part of marker value — `MKRVAL?`
- get secondary part of marker value — `MKRAUV?`
- get sweep parameter — `MKRPRM?`
- get data point number — `MKRP?`

You can get the sub-marker value using the following commands:

- get primary part of sub-marker value — `SMKRVAL{1-7}?`
- get secondary part of sub-marker value — `SMKRAUV{1-7}?`
- get sweep parameter — `SMKRPRM{1-7}?`
- get data point number — `SMKRP{1-7}?`

You can get the Δmarker value using the following commands:

- get primary part of Δmarker value — `DMKRVAL?`
- get secondary part of Δmarker value — `DMKRAUV?`
- get sweep parameter — `DMKRPRM?`

To Get Measurement Trace Using ASCII Format

```
10    !
20    ! Figure 4-2. To Get Measurement Trace Using ASCII Format
30    !
40    ASSIGN @Hp4396 TO 717  ! When iBASIC is used, change "717" to "800"
50    !
60    INPUT "ENTER CENTER FREQUENCY (Hz)",F_cent
70    INPUT "ENTER FREQUENCY SPAN (Hz)",F_span
80    OUTPUT @Hp4396;"CENT";F_cent
90    OUTPUT @Hp4396;"SPAN";F_span
100   !
110   OUTPUT @Hp4396;"CLES"
120   OUTPUT @Hp4396;"*SRE 4;ESNB 1"
130   ON INTR 7 GOTO Sweep_end  ! \ When iBASIC is used, change "7" to "8"
140   ENABLE INTR 7;2           ! /
150     OUTPUT @Hp4396;"SING"       ! Trigger a Measurement
160 Measuring:    GOTO Measuring    ! Measuring
170 Sweep_end:   !
180     DIM Dat(1:801,1:2),Swp(1:801)  ! For spectrum measurement, change
190     OUTPUT @Hp4396;"FORM4"         ! "Dat(1:801,1:2)" to "Dat(1:801)"
200     OUTPUT @Hp4396;"OUTPDTRC?"
210     ENTER @Hp4396 USING "%,K";Dat(*)
220     OUTPUT @Hp4396;"OUTPSWPRM?"
230     ENTER @Hp4396 USING "%,K";Swp(*)
240     !
250     OUTPUT @Hp4396;"POIN?"
260     ENTER @Hp4396;Nop
270     FOR I=1 TO Nop
280       PRINT Swp(I);"Hz",Dat(I,1);"dB" ! For spectrum measurement, change
290     NEXT I                            ! "Dat(I,1)" to "Dat(I)"
300     END
```

Figure 4-2. Sample Program : To Get Measurement Trace Using ASCII Format

Set the Receive Array

```
180     DIM Dat(1:801,1:2),Swp(1:801)
```

Line 180 sets the array size to the analyzer's maximum number of measurement points (801).

In this example, it is assumed that the analyzer is in the network analyzer mode of operation, in which each point has complex data. If you use the analyzer in the spectrum analyzer mode, each measurement point has only real data, so you must set the data array `Dat` as follows:

```
180     DIM Dat(1:801),Swp(1:801)

280     PRINT Swp(I);"Hz",Dat(I);"dB"
```

If the number of measurement points changes, then so does the number of data. You must control the number of entered measurement data (see lines 210 and 230).

Set Data Transfer Format

```
190     OUTPUT @Hp4396;"FORM4"
```

Line 190 tells the analyzer to use the ASCII transfer format.

Read Data

```
200     OUTPUT @Hp4396;"OUTPDTRC?"
210     ENTER @Hp4396 USING "%,K";Dat(*)
220     OUTPUT @Hp4396;"OUTPSWPRM?"
230     ENTER @Hp4396 USING "%,K";Swp(*)
```

`OUTPDTRC?` retrieves DATA TRACE ARRAYS, and `OUTPSWPRM?` retrieves sweep parameters.

In line 210 and 230, you must choose `%,K` to allow for an insufficient number of data points to fill the array (which is 801 as declared in line 180).

What are other data arrays?

You can retrieve the following data arrays, exchanging HP-IB command `OUTPDTRC?` in line 200. For details on each command, see the *HP-IB Command Reference* manual.

• RAW DATA ARRAYS	`OUTPRAW{1-4}?`
• DATA ARRAYS	`OUTPDATA?`
• MEMORY ARRAYS	`OUTPMEMO?`
• MEMORY TRACE ARRAYS	`OUTPMTRC?`
• CALIBRATION COEFFICIENT ARRAYS	`OUTPCALC{1-12}?`

To Get Measurement Trace Using Binary Format

Two programs are shown to get the data arrays using binary format in Figure 4-3 and Figure 4-5.

Figure 4-3	Figure 4-5
 For the external controller	 For the HP Instrument BASIC
• Using REDIM command (line 220) to allow for the change in the number of measurement points.	• HP Instrument BASIC allows for a different number of data points then the number specified in the data array declaration.

Before running the program in Figure 4-3, you must modify the dimension of the data arrays to match to the analyzer type (network or spectrum). (See the "Set the Receive Array" example.)

```
10  !
20  ! Figure 4-3. To Get Measurement Trace Using
30  !              IEEE 64-bit Floating point Format (For External Controller)
40  !
50    ASSIGN @Hp4396 TO 717
60    !
70    INPUT "ENTER CENTER FREQUENCY (Hz)",F_cent
80    INPUT "ENTER FREQUENCY SPAN (Hz)",F_span
90    OUTPUT @Hp4396;"CENT";F_cent
100   OUTPUT @Hp4396;"SPAN";F_span
110   !
120   OUTPUT @Hp4396;"CLES"
130   OUTPUT @Hp4396;"*SRE 4;ESNB 1"
140   ON INTR 7 GOTO Sweep_end   !
150   ENABLE INTR 7;2            !
160     OUTPUT @Hp4396;"SING"
170 Measuring:GOTO Measuring
180 Sweep_end:   !
190   DIM Dat(1:801,1:2),Swp(1:801)   ! For spectrum measurement, change
200   OUTPUT @Hp4396;"POIN?"          ! "Dat(1:801,1:2)" to "Dat(1:801)"
210   ENTER @Hp4396;Nop
220   REDIM Dat(1:Nop,1:2),Swp(1:Nop)
230   OUTPUT @Hp4396;"FORM3"
240   ASSIGN @Dt TO 717;FORMAT OFF
250   OUTPUT @Hp4396;"OUTPTDTRC?"
260   ENTER @Dt USING "%,8A";A$
270   ENTER @Dt;Dat(*)
280   ENTER @Dt USING "%,1A";B$
290   OUTPUT @Hp4396;"OUTPSWPRM?"
300   ENTER @Dt USING "%,8A";A$
310   ENTER @Dt;Swp(*)
320   ENTER @Dt USING "%,1A";B$
```

Figure 4-3.
Sample Program : To Get Measurement Trace Using IEEE 64-bit Floating Point Format (For External Controller) (1/2)

```
330   ASSIGN @Dt TO *
340   !
350   FOR I=1 TO Nop
360     PRINT Swp(I);"Hz",Dat(I,1);"dB"    ! For spectrum measurement, change
370   NEXT I                                ! "Dat(I,1)" to "Dat(I)"
380   END
```

Figure 4-3. Sample Program : To Get Measurement Trace Using IEEE 64-bit Floating Point Format (For External Controller) (2/2)

This program is similar to the ASCII transfer program. However, you must set the data transfer format OFF when using the binary data transfer format.

Set the Receive Array

```
190   DIM Dat(1:801,1:2),Swp(1:801)    ! For spectrum measurement, change
200   OUTPUT @Hp4396;"POIN?"           ! "Dat(1:801,1:2)" to "Dat(1:801)"
210   ENTER @Hp4396;Nop
220   REDIM Dat(1:Nop,1:2),Swp(1:Nop)
```

Line 190 sets the array size to the analyzer's maximum number of measurement points (801).

In this example, it is assumed that the analyzer is in the network analyzer mode of operation, in which each point has complex data. If you use the analyzer in the spectrum analyzer mode, each measurement point has only real data, so you must set the data array `Dat` as follows:

```
190     DIM Dat(1:801),Swp(1:801)

220     REDIM Dat(1:Nop),Swp(1:Nop)

360     PRINT Swp(I);"Hz",Dat(I);"dB"
```

Lines 200 and 210 interrogate the analyzer to determine the number of measurement points. Line 220 resizes the receive array to match the data.

Set Data Transfer Format

```
200   OUTPUT @Hp4396;"FORM3"
210   ASSIGN @Dt TO 717;FORMAT OFF ! When iBASIC is used, change "717" to "800"
```

To use FORM3 the computer must be instructed to stop formatting the incoming data with the `ENTER` statement. This is done by defining an I/O path with ASCII formatting OFF. The I/O path points to the analyzer. This path can be used to read or write data to the analyzer, as long as that data is in binary rather than ASCII format.

What are other binary data formats?

You can use the following data transfer formats, by changing the HP-IB command `FORM3` in line 200.

- IEEE 32 bit floating point format `FORM2`
- MS-DOS® personal computer format `FORM5`

Read Data

```
250     OUTPUT @Hp4396;"OUTPTDTRC?"
260     ENTER @Dt USING "%,8A";A$
270     ENTER @Dt;Dat(*)
280     ENTER @Dt USING "%,1A";B$
290     OUTPUT @Hp4396;"OUTPSWPRM?"
300     ENTER @Dt USING "%,8A";A$
310     ENTER @Dt;Swp(*)
320     ENTER @Dt USING "%,1A";B$
```

FORM3 has an eight-byte header to deal with. The first two bytes are the ASCII characters #6. This indicates that a fixed length block transfer follows and that the next 6 bytes form an integer specifying the number of bytes in the block to follow. The header must be read in so that data order is maintained (lines 260 and 300).

At the data end, the terminator "LF^EOI" is sent(lines 280 and 320).



Figure 4-4. FORM3 Data Transfer Format

Before running the program in Figure 4-5, you must modify the dimension of the data arrays to match the analyzer type (network or spectrum). (See the "Set the Receive Array" below.)

```
10  !
20  ! Figure 4-5. To Get Measurement Trace Using
30  !              IEEE 64-bit Floating point Format (For HP Instrument BASIC)
40  !
50    ASSIGN @Hp4396 TO 800
60    !
70    INPUT "ENTER CENTER FREQUENCY (Hz)",F_cent
80    INPUT "ENTER FREQUENCY SPAN (Hz)",F_span
90    OUTPUT @Hp4396;"CENT";F_cent
100   OUTPUT @Hp4396;"SPAN";F_span
110   !
120   OUTPUT @Hp4396;"CLES"
130   OUTPUT @Hp4396;"*SRE 4;ESNB 1"
140   ON INTR 8 GOTO Sweep_end
150   ENABLE INTR 8;2
160     OUTPUT @Hp4396;"SING"
170 Measuring:GOTO Measuring
180 Sweep_end:   !
190   DIM Dat(1:802,1:2),Swp(1:802)   ! For spectrum measurement, change
200   OUTPUT @Hp4396;"FORM3"          ! "Dat(1:802,1:2)" to "Dat(1:802)"
210   ASSIGN @Dt TO 800;FORMAT OFF
220   OUTPUT @Hp4396;"OUTPDTRC?"
230   ENTER @Dt USING "%,8A";A$
240   ENTER @Dt;Dat(*)
250   OUTPUT @Hp4396;"OUTPSWPRM?"
260   ENTER @Dt USING "%,8A";A$
270   ENTER @Dt;Swp(*)
280   ASSIGN @Dt TO *
290   !
300   OUTPUT @Hp4396;"POIN?"
310   ENTER @Hp4396;Nop
320   FOR I=1 TO Nop
330     PRINT Swp(I);"Hz",Dat(I,1);"dB"   ! For spectrum measurement, change
340   NEXT I                              ! "Dat(I,1)" to "Dat(I)"
350   END
```

Figure 4-5.
Sample Program : To Get Measurement Trace Using IEEE 64-bit Floating Point Format (For HP Instrument BASIC)

Set the Receive Array

```
190   DIM Dat(1:802,1:2),Swp(1:802)
```

Line 190 sets the array size to the maximum number of data. The analyzer's maximum number of measurement points is 801. At end of data, the terminator "LF^EOI" is sent (see Figure 4-4).

In this example, it is assumed that the analyzer is in the network analyzer mode of operation, in which each point has complex data. If you use the analyzer in the spectrum analyzer mode, each measurement point has only real data, so you must set the data array `Dat` as follows:

```
190     DIM Dat(1:802),Swp(1:802)
330     PRINT Swp(I);"Hz",Dat(I);"dB"
```

If the number of measurement points change, then so does the number of data. HP Instrument BASIC allows for insufficient data in lines 220 to 270 (see "Read Data").

Set Data Transfer Format

```
200   OUTPUT @Hp4396;"FORM3"
210   ASSIGN @Dt TO 800;FORMAT OFF
```

To use FORM3 the computer must be instructed to stop formatting the incoming data with the ENTER statement. This is the same operation that is done in the program in Figure 4-3. For more information, see "Set Data Transfer Format" in Figure 4-3.

Read Data

```
220     OUTPUT @Hp4396;"OUTPTDTRC?"
230     ENTER @Dt USING "%,8A";A$
240     ENTER @Dt;Dat(*)
250     OUTPUT @Hp4396;"OUTPSWPRM?"
260     ENTER @Dt USING "%,8A";A$
270     ENTER @Dt;Swp(*)
```

When all the data is transferred and no data is left for the receive arrays, `Dat(*)` and `Swp(*)`, the program goes to the next operation.

5

Writing Data Arrays to the Analyzer

Chapter 4 explained how to read data arrays from the analyzer. This chapter shows how to write data arrays to the analyzer. You can read, modify, and store the data arrays to the analyzer over the HP-IB. This allows you to modify the trace on the analyzer's display. For details on the data arrays in the analyzer, see the *HP-IB Command Reference* manual.

To Modify Calibration Data

```
10     !
20     ! Figure 5-1. To Modify Calibration Data
30     !
40     ASSIGN @Hp4396 TO 717  ! When iBASIC is used, change "717" to "800"
50     !
60     OUTPUT @Hp4396;"PRES"
70     OUTPUT @Hp4396;"NA"
80     INPUT "Enter center frequency(Hz).",F_cent
90     INPUT "Enter frequency span(Hz).",F_span
100    OUTPUT @Hp4396;"CENT ";F_cent
110    OUTPUT @Hp4396;"SPAN ";F_span
120    OUTPUT @Hp4396;"HOLD"
130    !
140    ! Calibration
150    OUTPUT @Hp4396;"CLES"
160    OUTPUT @Hp4396;"*SRE 4;ESNB 1"    ! Set enable STB and ESB
170    INPUT "Connect THRU and press [RETURN] to do CAL.",Dum$
180    OUTPUT @Hp4396;"CALI RESP"
190    ON INTR 7 GOTO Cal_end    ! \ When iBASIC is used,
200    ENABLE INTR 7;2           ! / change "7" to "8"
210      OUTPUT @Hp4396;"STANC"          ! Measure THRU
220 Calibrating:GOTO Calibrating
230 Cal_end:     !
240    OUTPUT @Hp4396;"RESPDONE"         ! Calculating cal coefficient
250    OUTPUT @Hp4396;"*OPC?"          ! \ Wait until calculating ends
260    ENTER @Hp4396;Dum               ! /
270    DISP "Calibration Complete"
280    !
290    !  Read Calibration Data
300    DIM Dat(1:801,1:2)   ! When iBASIC is used, change "801" to "802"
310    OUTPUT @Hp4396;"POIN?"  ! \
320    ENTER @Hp4396;Nop       ! | When iBASIC is used,delete these lines
330    REDIM Dat(1:Nop,1:2)    ! /
340    ASSIGN @Dt TO 717;FORMAT OFF     ! When iBASIC is used,
350    OUTPUT @Hp4396;"FORM3"           ! change "717" to "800"
```

Figure 5-1. Sample Program : To Modify Calibration Data (1/2)

```
360     OUTPUT @Hp4396;"OUTPCALC1?"
370     ENTER @Dt USING "%,8A";Head$
380     ENTER @Dt;Dat(*)
390     ENTER @Dt USING "%,1A";Dum$  ! When iBASIC is used, delete this line
400 !
410 !    Modify Calibration Data
420 !
430     !   Restore Calibration  Data
440     OUTPUT @Hp4396;"INPUCALC1 ";
450     OUTPUT @Dt USING "#,8A";Head$
460     OUTPUT @Dt;Dat(*),END
470     ASSIGN @Dt TO *
480     OUTPUT @Hp4396;"SAVC" ! Redraw Trace
490     END
```

Figure 5-1. Sample Program : To Modify Calibration Data (2/2)

This program measures calibration standards, reads the obtained calibration data, and restores the data in the analyzer. For details on how to obtain the calibration data by measuring the standard, see Chapter 1. For details on how to transfer the data arrays, see Chapter 4.

Read Calibration Data

```
290     !  Read Calibration Data
300     DIM Dat(1:801,1:2)   ! When iBASIC is used, change "801" to "802"
310     OUTPUT @Hp4396;"POIN?"  ! \
320     ENTER @Hp4396;Nop       ! | When iBASIC is used,delete these lines
330     REDIM Dat(1:Nop,1:2)    ! /
340     ASSIGN @Dt TO 717;FORMAT OFF     ! When iBASIC is used,
350     OUTPUT @Hp4396;"FORM3"           ! change "717" to "800"
360     OUTPUT @Hp4396;"OUTPCALC1?"
370     ENTER @Dt USING "%,8A";Head$
380     ENTER @Dt;Dat(*)
390     ENTER @Dt USING "%,1A";Dum$  ! When iBASIC is used, delete this line
```

The controller can read out the error coefficients using the HP-IB commands `OUTPCALC{1-12}`. Each point is a real/imaginary pair, and the number of points in the array is the same as the number of points in the sweep. For details on data transfer, see Chapter 4.

Each calibration type uses only as many arrays as needed, starting with array 1, and each array stores a specific error coefficient. Therefore, it is necessary to know the type of calibration about to be read out: attempting to read an array not being used in the current calibration causes the "REQUESTED DATA NOT CURRENTLY AVAILABLE" warning to be displayed. For assignment of data arrays, see the *HP-IB Command Reference* manual.

Modify Calibration Data

```
400 !
410 !   Modify Calibration Data
420 !
```

In this portion of program, you modify the CALIBRATION COEFFICIENT ARRAY, which is contained in `Dat(1:801,1:2)`.

Restore Modified Calibration Data

```
430    !   Restore Calibration  Data
440    OUTPUT @Hp4396;"INPUCALC1 ";
450    OUTPUT @Dt USING "#,8A";Head$
460    OUTPUT @Dt;Dat(*),END
```

Line 440 opens the CALIBRATION COEFFICIENT ARRAY 1 in the analyzer. This array is used to restore the data.

Lines 450 and 460 send the file header (`Head$`), calibration data (`Dat(*)`) and the terminator (`END`). The file header is an input in line 370.

This example sets the trigger to HOLD at line 120. The analyzer does not redraw the trace with the new CALIBRATION COEFFICIENT ARRAYS when the trigger is set to HOLD. You can redraw the trace by issuing the HP-IB command `SAVC`. For details, see the "Redrawing Measurement Trace with Modified Calibration Data" description.

What is a file header? When using the binary data transfer format, the transferred data must be accompanied by the file header that represents the data length. In this example, the data transfer format is FORM3 and the transferred data is configured as follows:



FORM3 Data Transfer Format

If you are not reading the header, you can create it using the number of data points. Change the program lines 440 to 460 as follows:

```
440    OUTPUT @Hp4396;"POIN?"
441    ENTER @Hp4396;Nop
442    V$=VAL$(Nop*2*8)
443    Numv=LEN(V$)
444    Head$="000000"
445    FOR I=1 TO Numv
446      Head$[7-I,7-I]=V$[Numv-I+1,Numv-I+1]
447    NEXT I
448    !
449    OUTPUT @Hp4396;"INPUCALC1 ";
450    OUTPUT @Dt USING "#,8A";"#6"&Head$
460    OUTPUT @Dt;Dat(*),END
```

Lines 440 to 442 calculate the number of bytes transferred (8 byte for real part, 8 byte for imaginary part), and represents it in the string format.

Line 443 counts the number of characters in the string that contains the number of bytes transferred.

Line 444 enters 0 as the initial value in all header arrays.

Lines 445 to 447 place the number of bytes transferred to the header array digit by digit from the sixth array to the first array of the header.

For example, if the number of points is 201, the value of `Head$` is 003216.

Redrawing Measurement Trace with Modified Calibration Data

```
480     OUTPUT @Hp4396;"SAVC" ! Redraw Trace
```

When all the calibration coefficients are in the analyzer, send the HP-IB command SAVC to have the analyzer reshape a trace with the coefficients.

How can you modify the trace? (Summary) To modify the trace on the display, you rewrite the data arrays in the analyzer. Figure 5-2 shows the relation of the data arrays, data processing, and HP-IB commands.



Figure 5-2. Data Arrays, Data Processing, and HP-IB Command

- Reset command

*RST or PRES command clears all arrays.

- Data array writing command

INPURAW{1-4}, INPUDATA and INPUDTRC commands write the corresponding arrays. These commands immediately reshape the data trace on the analyzer's display.

INPUCALC{1-12} commands write the CALIBRATION COEFFICIENT ARRAYS.

- DATA to MEMORY command

DATMEM command restores the contents in DATA ARRAYS into MEMORY ARRAYS, and the contents in DATA TRACE ARRAYS into MEMORY TRACE ARRAYS.

- Data processing command

SAVC command executes the data processing CORRECTION with the current RAW ARRAYS and CALIBRATION COEFFICIENT ARRAYS.

The following examples show how to modify the DATA ARRAYS and DATA TRACE ARRAYS.

To Modify Error-Corrected Data

```
10    !
20    ! Figure 5-3. To Modify Error-Corrected Data
30    !
40    ASSIGN @Hp4396 TO 717  ! When iBASIC is used, change "717" to "800"
50    !
60    OUTPUT @Hp4396;"PRES"
70    OUTPUT @Hp4396;"NA"
80    INPUT "Enter center frequency(Hz).",F_cent
90    INPUT "Enter frequency span(Hz).",F_span
100   OUTPUT @Hp4396;"CENT ";F_cent
110   OUTPUT @Hp4396;"SPAN ";F_span
120    !
130   INPUT "Connect DUT and press [RETURN].",Dum$
140   OUTPUT @Hp4396;"CLES"
150   OUTPUT @Hp4396;"*SRE 4;ESNB 1"    ! Set enable STB and ESB
160   ON INTR 7 GOTO Sweep_end  ! \ When iBASIC is used,
170   ENABLE INTR 7;2           ! / change "7" to "8"
180     OUTPUT @Hp4396;"SING"
190 Measuring:     GOTO Measuring
200 Sweep_end:     !
210   DISP "Measurement Complete"
220   !
230   !  Read Error-Corrected Data
240   DIM Dat(1:801,1:2)   ! When iBASIC is used, change "801" to "802"
250   OUTPUT @Hp4396;"POIN?"  ! \
260   ENTER @Hp4396;Nop       ! | When iBASIC is used, delete these lines
270   REDIM Dat(1:Nop,1:2)    ! /
280   ASSIGN @Dt TO 717;FORMAT OFF   ! When iBASIC is used,
290   OUTPUT @Hp4396;"FORM3"         ! change "717" to "800"
300   OUTPUT @Hp4396;"OUTPDATA?"
310   ENTER @Dt USING "%,8A";Head$
320   ENTER @Dt;Dat(*)
330   ENTER @Dt USING "%,1A";Dum$    ! When iBASIC is used, delete this line
340 !
350 !   Modify Error-Corrected Data
360 !
370   !   Restore Error-Corrected Data
380   OUTPUT @Hp4396;"INPUDATA ";
390   OUTPUT @Dt USING "#,8A";Head$
400   OUTPUT @Dt;Dat(*),END
410   ASSIGN @Dt TO *
420   END
```

Figure 5-3. Sample Program : To Modify Error-Corrected Data

This program measures the DUT, reads the obtained data, and restores the data in the analyzer. For details on how to read the data array, see Chapter 4. For information on how to modify the trace on the display, see the "To Modify Calibration Data" example.

Read Error-Corrected Data

```
230   !  Read Error-Corrected Data
240   DIM Dat(1:801,1:2)   ! When iBASIC is used, change "801" to "802"
250   OUTPUT @Hp4396;"POIN?"  ! \
260   ENTER @Hp4396;Nop       ! | When iBASIC is used, delete these lines
270   REDIM Dat(1:Nop,1:2)    ! /
280   ASSIGN @Dt TO 717;FORMAT OFF   ! When iBASIC is used,
290   OUTPUT @Hp4396;"FORM3"         ! change "717" to "800"
300   OUTPUT @Hp4396;"OUTPDATA?"
310   ENTER @Dt USING "%,8A";Head$
320   ENTER @Dt;Dat(*)
330   ENTER @Dt USING "%,1A";Dum$    ! When iBASIC is used, delete this line
```

OUTPDATA? command (line 300) retrieves DATA ARRAYS in the analyzer. For details on data transfer, see Chapter 4.

Restore Modified Error-Corrected Data

```
370   !   Restore Error-Corrected Data
380   OUTPUT @Hp4396;"INPUDATA ";
390   OUTPUT @Dt USING "#,8A";Head$
400   OUTPUT @Dt;Dat(*),END
```

Line 380 opens the DATA ARRAYS in the analyzer to restore the data.

Lines 390 to 400 transfer data in FORM3 (a similar procedure is used in the "To Modify Calibration Data" example).

To Modify Trace Data

```
10    !
20    ! Figure 5-4. To Modify Trace Data
30    !
40    ASSIGN @Hp4396 TO 717  ! When iBASIC is used, change "717" to "800"
50    !
60    OUTPUT @Hp4396;"PRES"
70    OUTPUT @Hp4396;"NA"
80    INPUT "Enter center frequency(Hz).",F_cent
90    INPUT "Enter frequency span(Hz).",F_span
100   OUTPUT @Hp4396;"CENT ";F_cent
110   OUTPUT @Hp4396;"SPAN ";F_span
120    !
130   INPUT "Connect DUT and press [RETURN].",Dum$
140   OUTPUT @Hp4396;"CLES"
150   OUTPUT @Hp4396;"*SRE 4;ESNB 1"    ! Set enable STB and ESB
160   ON INTR 7 GOTO Sweep_end    ! \ When iBASIC is used,
170   ENABLE INTR 7;2             ! / change "7" to "8"
180     OUTPUT @Hp4396;"SING"
190 Measuring:     GOTO Measuring
200 Sweep_end:     !
210   DISP "Measurement Complete"
220   !
230   !  Read Trace Data
240   DIM Dat(1:801,1:2)  ! When iBASIC is used, change "801" to "802"
250   OUTPUT @Hp4396;"POIN?"  ! \
260   ENTER @Hp4396;Nop       ! | When iBASIC is used, delete these lines
270   REDIM Dat(1:Nop,1:2)    ! /
280   ASSIGN @Dt TO 717;FORMAT OFF   ! When iBASIC is used,
290   OUTPUT @Hp4396;"FORM3"         ! change "717" to "800"
300   OUTPUT @Hp4396;"OUTPDTRC?"
310   ENTER @Dt USING "%,8A";Head$
320   ENTER @Dt;Dat(*)
330   ENTER @Dt USING "%,1A";Dum$  ! When iBASIC is used, delete this line
340 !
350 !   Modify Trace Data
360 !
370   !   Restore Trace Data
380   OUTPUT @Hp4396;"INPUDTRC ";
390   OUTPUT @Dt USING "#,8A";Head$
400   OUTPUT @Dt;Dat(*),END
410   ASSIGN @Dt TO *
420   END
```

Figure 5-4. Sample Program : To Modify Trace Data

This program measures the DUT, reads the obtained data, and restores the data into the analyzer. For details on how to read the data array, see Chapter 4.

For details on how to modify the trace on the display, see the "To Modify Calibration Data" example.

Read Trace Data

```
230   !  Read Trace Data
240   DIM Dat(1:801,1:2)  ! When iBASIC is used, change "801" to "802"
250   OUTPUT @Hp4396;"POIN?"  ! \
260   ENTER @Hp4396;Nop       ! | When iBASIC is used, delete these lines
270   REDIM Dat(1:Nop,1:2)    ! /
280   ASSIGN @Dt TO 717;FORMAT OFF   ! When iBASIC is used,
290   OUTPUT @Hp4396;"FORM3"         ! change "717" to "800"
300   OUTPUT @Hp4396;"OUTPDTRC?"
310   ENTER @Dt USING "%,8A";Head$
320   ENTER @Dt;Dat(*)
330   ENTER @Dt USING "%,1A";Dum$  ! When iBASIC is used, delete this line
```

The OUTPDTRC? command (line 300) retrieves trace data in the analyzer. For details on data transfer, see Chapter 4.

Restore Modified Trace Data

```
370   !   Restore Trace Data
380   OUTPUT @Hp4396;"INPUDTRC ";
390   OUTPUT @Dt USING "#,8A";Head$
400   OUTPUT @Dt;Dat(*),END
```

Line 380 opens the DATA TRACE ARRAYS in the analyzer to restore the data.

Lines 390 and 400 transfer data in FORM3 (a similar procedure is used in the "To Modify Calibration Data" example).

6

Printing or Plotting the Analyzer's Display

This chapter describes how to print or plot the information on the analyzer display using HP-IB commands.

The analyzer has two HP-IB operating modes that can only be selected using the (LOCAL) menu's `SYSTEM CONTROLLER` and `ADDRESSABLE ONLY` softkeys. There is no HP-IB command to choose the system controller mode, you can only select it from the front panel.



Using HP Instrument BASIC (Option 1C2)

Set the analyzer to `SYSTEM CONTROLLER`. Initially, the analyzer is the active controller and you can execute print or plot commands immediately.



Using computer for system controller

Set the analyzer to `ADDRESSABLE ONLY`. To print or plot the displayed information, it is necessary to pass active control to the analyzer.

Printer/Plotter Preparation

1. Connect a printer/plotter using an HP-IB cable.
2. Set the printer/plotter's HP-IB address.

   Typically,

   Printer 1

   Plotter 5
3. Turn the printer/plotter on.

To Print Analyzer Display

```
10    !
20    ! Figure 6-1. To Print Analyzer Display
30    !
40    ASSIGN @Hp4396 TO 717  ! When iBASIC is used, change "717" to "800"
50    !
60    !   Measure DUT, and display the measurement
70    !   trace to be printed.
80    !
90    OUTPUT @Hp4396;"ADDRPRIN 1"
100   OUTPUT @Hp4396;"CLES"      ! \ When iBASIC is used,
110   OUTPUT @Hp4396;"*ESE 2"    ! / delete these lines
120   !
130   OUTPUT @Hp4396;"PRINALL"
140   REPEAT                     ! \
150     Stat=SPOLL(@Hp4396)      !  |
160   UNTIL BIT(Stat,5)          !  |
170                              !  |
180   PASS CONTROL @Hp4396       !  | When iBASIC is used,
190   DISP "PRINTING"            !  | delete these lines
200   REPEAT                     !  |
210    STATUS 7,6;Hpib           !  |
220   UNTIL BIT(Hpib,6)          !  |
230   DISP "DONE"                ! /
240   END
```

Figure 6-1. Sample Program : To Print the Analyzer Display

Enter Printer's Address to Analyzer

```
90    OUTPUT @Hp4396;"ADDRPRIN 1"
```

Enters the printer's HP-IB address (typically 1).

Execute Print

```
130   OUTPUT @Hp4396;"PRINALL"
```

The PRINALL command copies the measurement display according to the plotting option.

What is the plotting option?

You can select the information to be copied to the printer with the following command.

```
PSOFT ON|OFF, PRIS, PRIC, PLOC
```

These commands must be executed before the PRINALL command. For details about these commands, see the *HP-IB Command Reference* manual.

Pass Control



When using HP Instrument BASIC, the pass control operation is not required, therefore, you must delete (or comment out) lines 100 and 110, and lines 140 to 230.

```
100   OUTPUT @Hp4396;"CLES"       ! \ When iBASIC is used,
110   OUTPUT @Hp4396;"*ESE 2"     ! / delete these lines
120   !
130   OUTPUT @Hp4396;"PRINALL"
140   REPEAT                      ! \
150     Stat=SPOLL(@Hp4396)       !  |
160   UNTIL BIT(Stat,5)           !  |
170                               !  |
180   PASS CONTROL @Hp4396        !  | When iBASIC is used,
190   DISP "PRINTING"             !  | delete these lines
200   REPEAT                      !  |
210    STATUS 7,6;Hpib            !  |
220   UNTIL BIT(Hpib,6)           !  |
230   DISP "DONE"                 ! /
```

Lines 100 and 110 allow Status Byte Register bit 5 to be set when the analyzer requests control.



If the analyzer is set to `ADDRESSABLE ONLY` and receives an HP-IB command telling it to plot or print the display (line 130), it sets the Request Control bit (bit 1 of the Event Status Register).

From line 140 to 160, the computer waits for the Request Control bit to set the Event Status Register bit, it then passes control (line 180) to the analyzer. The analyzer takes control of the bus, accesses the peripheral, and starts to print.

When the `PRINALL` command completes printing and the analyzer no longer needs control, it passes control back to the computer. From line 200 to 220, the computer is waiting for active control to be passed back.

For advanced pass control techniques, see Chapter 7.

To Plot Analyzer Display

```
10    !
20    ! Figure 6-2. To Plot Analyzer Display
30    !
40    ASSIGN @Hp4396 TO 717   ! When iBASIC is used, change "717" to "800"
50    !
60    !   Measure DUT, and display the measurement
70    !   trace to be plotted.
80    !
90    OUTPUT @Hp4396;"ADDRPLOT 5"
100   OUTPUT @Hp4396;"CLES"      ! \ When iBASIC is used,
110   OUTPUT @Hp4396;"*ESE 2"    ! / delete these lines.
120   !
130   OUTPUT @Hp4396;"PLOT"
140   REPEAT                     ! \
150     Stat=SPOLL(@Hp4396)      ! |
160   UNTIL BIT(Stat,5)          ! |
170                              ! |
180   PASS CONTROL @Hp4396       ! | When iBASIC is used,
190   DISP "PLOTTING"            ! | delete these lines
200   REPEAT                     ! |
210     STATUS 7,6;Hpib          ! |
220   UNTIL BIT(Hpib,6)          ! |
230   DISP "DONE"                ! /
240   !
250   END
```

Figure 6-2. Sample Program : To Plot Analyzer Display

This program is similar to printing the display. The analyzer requires active control to plot. For details of how to pass control, see the "To Print Analyzer Display" example.

Enter Plotter's Address to the Analyzer

```
90    OUTPUT @Hp4396;"ADDRPLOT 5"
```

Enter the plotter's address (typically 5).

Execute Plot

```
130   OUTPUT @Hp4396;"PLOT"
```

The PLOT command plots the displayed information according to the plotting option. For details of the plotting option, see the "To Print Analyzer Display" example.

7

Controlling HP Instrument BASIC from Remote

This chapter is for programmers who use both HP Instrument BASIC and an external controller at the same time. This chapter shows how to pass control and how to use the PROGram subsystem commands. The PROGram subsystem is part of the SCPI commands hierarchy. See the *HP-IB Command Reference* manual for more information on how to use SCPI commands.

If your HP-IB has only one controller (either an external computer or the HP Instrument BASIC), skip this chapter.

To Control HP-IB from HP Instrument BASIC

The analyzer must be the active controller of the HP-IB (select code 7) when controlling the device on the HP-IB (for example, a printer or plotter). Initially, the system controller is active. The active controller can pass control to the analyzer (or other controllers, if there are others on the HP-IB).

To print or plot using HP Instrument BASIC, the external controller must pass control to the analyzer.



Figure 7-1. Pass Control

What is an active controller?

- The active controller determines which controller can manage the HP-IB (that is, have control). The HP-IB can have only one active controller at a time. If two or more controllers are on the same bus, control is passed from one controller to another.
- Resetting the HP-IB (this operation can only be done by the system controller), the returns control to the system controller.

What is system controller?

- The controller that acts as the master controller.
- A controller can be set as either a system controller or a non-system controller. For details on how to set a controller, see the controller's manual. The analyzer, as a controller, can be used in either `SYSTEM CONTROLLER` (system controller) or `ADDRESSABLE ONLY` (non-system controller) mode. However, there can be only one system controller on the bus.

In this guide, it is assumed that the external controller is the system controller and the analyzer is used in `ADDRESSABLE ONLY` mode.

```
10  !
20  ! Figure 7-2. To Receive Control (on HP Instrument BASIC)
30  !
40  PRINTER IS 701
50  ON ERROR GOTO Not_active
60  Not_active: !
70  PRINT "HELLO WORLD!"
80  !
90  OFF ERROR
100 END
```

Figure 7-2. Sample Program: To Receive Control (On HP Instrument BASIC)

In line 60, to print out to the printer at address 701, the analyzer requires active control. Therefore, until active control is passed to the analyzer, the program loops back to line 30. When control is passed to the analyzer, it executes line 70.

To pass active control to HP Instrument BASIC:

```
PASS CONTROL 717 [Return]
```

Pass Control (On External Controller)

While the analyzer has control, it is free to address devices to talk and listen as needed. As the active controller, the analyzer can send messages to and read replies back from printers and plotters.

Note

The ability to assert the HP-IB interface clear line (IFC) and remote enable line (REN) are reserved for the system controller. Even when HP Instrument BASIC has active control, it is denied these functions.

`ABORT 7`	*assert the interface clear line (IFC)*
`REMOTE 7`	*assert the remote enable line (REN)*

To return active control to the system controller:

```
PASS CONTROL 721 [Return]
```

Return Control (On HP Instrument BASIC)

Or, you can return control to the external controller by resetting the HP-IB as follows:

```
ABORT 7 (Return)
```

Return Control (On External Controller)

This returns active control to the system controller.

Note

The PROGram subsystem commands in the following programs can be used on the external controller. See the *HP-IB Command Reference* manual for additional information on the use of SCPI commands.

To Execute an HP Instrument BASIC Command from the External Controller

Combine the `PROG:EXEC` command with HP Instrument BASIC command to be executed. For example, to execute `EDIT` command,

```
OUTPUT 717;"PROG:EXEC ""EDIT"""
```

Or you can change the double quoted term, `""EDIT""` to a single quote `'EDIT'`, as follows.

```
OUTPUT 717;"PROG:EXEC 'EDIT'"
```

Be careful when you execute an HP Instrument BASIC command that requires a parameter. For example, to execute the HP Instrument BASIC command `"GET "FILENAME""`, the quotation is as follows.

```
OUTPUT 717;"PROG:EXEC ""GET """"FILENAME""""""""
```

To Run an HP Instrument BASIC Program From the External Controller

```
10 ! Figure 7-3. To Run iBASIC Program From Ext.Controller
20 !             (On External Controller)
30 !
40   ABORT 7
50   ASSIGN @Hp4396 TO 717
60   OUTPUT @Hp4396;"PROG:DEL:ALL"
70   OUTPUT @Hp4396;"PROG:DEF #0"
80   OUTPUT @Hp4396;"10 MSI "":INTERNAL"""
90   OUTPUT @Hp4396;"20 GET ""FIG3_3"""
100  OUTPUT @Hp4396;"30 END"
110  OUTPUT @Hp4396;" " END
120  OUTPUT @Hp4396;"PROG:EXEC ""RUN"""
130  END
```

Figure 7-3.
Sample Program : To Run the HP Instrument BASIC Program From the External Controller (On External Controller)

You must put the disk that contains `FIG3_3` into the built-in disk drive of the analyzer before running the program. (The sample program disk for HP Instrument BASIC contains `FIG3_3`.)

Open the HP Instrument BASIC Editor

```
60    OUTPUT @Hp4396;"PROG:DEL:ALL"
70    OUTPUT @Hp4396;"PROG:DEF #0"
```

Scratch any program currently existing in the analyzer's HP Instrument BASIC editor and open the editor.

Send the HP Instrument BASIC Program

```
80    OUTPUT @Hp4396;"10 MSI "":INTERNAL"""
90    OUTPUT @Hp4396;"20 GET ""FIG3_3"""
100   OUTPUT @Hp4396;"30 END"
```

In the HP Instrument BASIC editor, the following program is now present:

```
10 MSI ":INTERNAL"
20 GET "FIG3_3"
30 END
```

Close the HP Instrument BASIC Editor

```
110   OUTPUT @Hp4396;" " END
```

Sending the `END` command to the analyzer closes the editor.

Run the Instrument BASIC Program

```
120    OUTPUT @Hp4396;"PROG:EXEC ""RUN"""
```

Line 120 runs the following program in the HP Instrument BASIC editor:

```
10 MSI ":INTERNAL"
20 GET "FIG3_3"
30 END
```

Line 20 retrieves the FIG3_3 program and at the same time runs the program.

To Transfer the Program to HP Instrument BASIC

```
10 !
20 !  Figure 7-4. To Transfer the Program to iBASIC (on External Controller)
30 !
40     ABORT 7
50     ASSIGN @Hp4396 TO 717
60     INPUT "FILENAME?",File_name$
70     OUTPUT @Hp4396;"PROG:DEL:ALL"
80     OUTPUT @Hp4396;"PROG:DEF #0"
90     ASSIGN @File TO File_name$
100    ON ERROR GOTO Done
110      DIM Line$[1024]
120      LOOP
130        Line$=""
140        ENTER @File;Line$
150        OUTPUT @Hp4396;Line$
160      END LOOP
170    Done:  !
180        OFF ERROR
190        OUTPUT @Hp4396;" " END
200        END
```

Figure 7-4.
Sample Program : To Transfer the Program to HP Instrument BASIC (on External Controller)

This Program transfers the program file in the mass storage of the external controller.

Before you run this program, confirm that the file to be transferred is on the mass storage device.

Open the HP Instrument BASIC Editor

```
70     OUTPUT @Hp4396;"PROG:DEL:ALL"
80     OUTPUT @Hp4396;"PROG:DEF #0"
```

Scratch any program that currently exists in the analyzer's HP Instrument BASIC editor and open the editor.

Transfer the Program

```
90    ASSIGN @File TO File_name$
100   ON ERROR GOTO Done
110     DIM Line$[1024]
120     LOOP
130       Line$=""
140       ENTER @File;Line$
150       OUTPUT @Hp4396;Line$
160     END LOOP
```

Transfer the program by line to the analyzer. When all program lines are transferred, the computer exits the loop and goes to line 170.

Close the HP Instrument BASIC Editor

```
190       OUTPUT @Hp4396;" " END
```

Sending the analyzer the `END` command closes the editor.

To Load an Array in an HP Instrument BASIC Program to the External Controller

```
10  !
20  ! Figure 7-5. To Load iBASIC Program Array (on External Controller)
30  !
40    ABORT 7
50    ASSIGN @Hp4396 TO 717
60    DIM Passed(1:801,1:2)
70    OUTPUT @Hp4396;"PROG:NUMB? ""Dat"";"
80    ENTER @Hp4396;Passed(*)
90    END
```

Figure 7-5.
Sample Program : To Load HP Instrument BASIC Program Array (on External Controller)

This program retrieves the array generated in the sample program listed in Figure 4-2 when that program is executed in HP Instrument BASIC. This information is transferred to the external controller.

First, run the HP Instrument BASIC program `FIG4_2` to store the data into the array. Then run the program in Figure 7-5 to transfer the data.

Transfer the Program Array of HP Instrument BASIC

```
70    OUTPUT @Hp4396;"PROG:NUMB? ""Dat"";"
80    ENTER @Hp4396;Passed(*)
```

The `PROG:NUMB? "Dat"` query returns the program array `Dat(1:801,1:2)` of Figure 4-2. The array is entered into `Passed(1:801,1:2)`.

8

Programming Limit Test from Remote

This chapter describes how to perform limit tests, set the limit lines, test the DUT, and transfer the results of the test.

To Perform limit Test

```
10   !
20   ! Figure 8-1. Limit Test
30   !
40    ASSIGN @Hp4396 TO 717  ! When iBASIC is used, change "717" to "800"
50   !
60    CLEAR SCREEN
70    PRINT USING "10A,15A,15A,15A";"Segment","Swp.Prmtr(Hz)","Upper","Lower"
80   !
90    DIM Table(1:18,1:3)
100   INPUT "Enter number of segments (<=18)",Numb
110   FOR I=1 TO Numb
120     GOSUB Loadlimit
130   NEXT I
140   !
150   LOOP
160      INPUT "Do you want to edit? (Y/N)",An$
170   EXIT IF An$="N" OR An$="n"
180      INPUT "Enter segment number(<=18)",I
190   IF Numb<I THEN Numb=I
200      GOSUB Loadlimit
210   END LOOP
220   !
230   OUTPUT @Hp4396;"EDITLIML"
240   OUTPUT @Hp4396;"LIMCLEL"
250   FOR K=1 TO Numb
260     OUTPUT @Hp4396;"LIMSADD"
270     OUTPUT @Hp4396;";LIMPRM ";Table(K,1)
280     OUTPUT @Hp4396;";LIMU ";Table(K,2)
290     OUTPUT @Hp4396;"LIML ";Table(K,3)
300     OUTPUT @Hp4396;"LIMSDON"
310   NEXT K
320   OUTPUT @Hp4396;"LIMEDONE"
330   OUTPUT @Hp4396;"LIMILINE ON"
340 !
350   INPUT "Connect DUT, and press Enter.",Dum$
```

Figure 8-1. Sample Program : Limit Test (1/2)

```
360   OUTPUT @Hp4396;"CLES"
370   OUTPUT @Hp4396;"*SRE 4;ESNB 1"
380   ON INTR 7 GOTO Sweep_end  ! \ When iBASIC is used, change "7" to "8"
390   ENABLE INTR 7;2           ! /
400     OUTPUT @Hp4396;"SING"
410 Measuring:GOTO Measuring
420 Sweep_end:   !
430   OUTPUT @Hp4396;"LIMITEST ON"
440   DIM Dt(1:801,1:4)
450   OUTPUT @Hp4396;"OUTPLIMF?"         ! \ Output test results.
460   ENTER @Hp4396 USING "%,K";Dt(*)    ! /
470   OUTPUT @Hp4396;"OUTPFAIP?"
480   ENTER @Hp4396;Failp
490   IF Failp=0 THEN Passed
500   PRINT "           FAIL POINTS              "
510   FOR I=1 TO Failp
520     PRINT
530     PRINT "Swp. prmtr : ";Dt(I,1)
540     PRINT "   Results       Upper         Lower    "
550     PRINT TAB(5);Dt(I,2);TAB(17);Dt(I,3);TAB(32);Dt(I,4)
560   NEXT I
570 Passed: !
580   DISP "Program End"
590   STOP
600   !
610 Loadlimit: !
620   INPUT "ENTER SWEEP PARAMETER (Hz)",Table(I,1)
630   INPUT "ENTER UPPER LIMIT VALUE",Table(I,2)
640   INPUT "ENTER LOWER LIMIT VALUE",Table(I,3)
650   PRINT I;TAB(11);Table(I,1);TAB(27);Table(I,2);TAB(42);Table(I,3)
660   RETURN
670 END
```

Figure 8-1. Sample Program : Limit Test (2/2)

Edit Limit Line

```
60   CLEAR SCREEN
70   PRINT USING "10A,15A,15A,15A";"Segment","Swp.Prmtr(Hz)","Upper","Lower"
80   !
90   DIM Table(1:18,1:3)
100   INPUT "Enter number of segments (<=18)",Numb
110   FOR I=1 TO Numb
120     GOSUB Loadlimit
130   NEXT I
140   !
150   LOOP
160      INPUT "Do you want to edit? (Y/N)",An$
170   EXIT IF An$="N" OR An$="n"
180      INPUT "Enter segment number(<=18)",I
190   IF Numb<I THEN Numb=I
200      GOSUB Loadlimit
210   END LOOP
```

```
  :
610 Loadlimit: !
620   INPUT "ENTER SWEEP PARAMETER (Hz)",Table(I,1)
630   INPUT "ENTER UPPER LIMIT VALUE",Table(I,2)
640   INPUT "ENTER LOWER LIMIT VALUE",Table(I,3)
650   PRINT I;TAB(11);Table(I,1);TAB(27);Table(I,2);TAB(42);Table(I,3)
660   RETURN
```

Lines 60 and 70 print the limit table heads on the BASIC SCREEN.

Line 90 defines the table (array `Table(1:18,1:3)`) used to hold the limit values. It contains the sweep parameter, the upper limit value, and the lower limit value as follows:

Segment	Sweep Parameter	Upper Limit	Lower Limit
1	Table(1,1)	Table(1,2)	Table(1,3)
2	Table(2,1)	Table(2,2)	Table(2,3)
⋮	⋮	⋮	⋮

Lines 110 to 130 call the subroutine `Loadlimit` (line 610) to edit and print as many segments as you defined in line 100 (the analyzer can retain up to 18 segments).

```
Segment   Swp.Prmtr(Hz)  Upper        Lower
 1         2.E+6          0           -10
 2         3.E+6          10          -20
 3         4.E+6          10          -10
```

The loop, lines 150 to 210, determines if you want to edit the table and confirms that the segment is in the table.

Set Limit Line

```
230   OUTPUT @Hp4396;"EDITLIML"                  Start to set the limit line
240   OUTPUT @Hp4396;"LIMCLEL"                   Clear existing limit table
250   FOR K=1 TO Numb
260     OUTPUT @Hp4396;"LIMSADD"                 Start to set a segment
270     OUTPUT @Hp4396;";LIMPRM ";Table(K,1)     Enter sweep parameter
280     OUTPUT @Hp4396;";LIMU ";Table(K,2)       Enter upper limit
290     OUTPUT @Hp4396;"LIML ";Table(K,3)        Enter lower limit
300     OUTPUT @Hp4396;"LIMSDON"                 Complete the segment
310   NEXT K
320   OUTPUT @Hp4396;"LIMEDONE"                  Complete the limit line
330   OUTPUT @Hp4396;"LIMILINE ON"               Set limit line ON
```



In this portion of the program, the limit table (edited using BASIC) is transferred to the analyzer.

Read the Limit Test Results

```
430   OUTPUT @Hp4396;"LIMITEST ON"
440   DIM Dt(1:801,1:4)
450   OUTPUT @Hp4396;"OUTPLIMF?"          ! \ Output test results.
460   ENTER @Hp4396 USING "%,K";Dt(*)     ! /
470   OUTPUT @Hp4396;"OUTPFAIP?"
480   ENTER @Hp4396;Failp
490   IF Failp=0 THEN Passed
500   PRINT "            FAIL POINTS              "
510   FOR I=1 TO Failp
520     PRINT
530     PRINT "Swp. prmtr : ";Dt(I,1)
540     PRINT "   Results       Upper         Lower    "
550     PRINT TAB(5);Dt(I,2);TAB(17);Dt(I,3);TAB(32);Dt(I,4)
560   NEXT I
570 Passed: !
580   DISP "Program End"
```

The OUTPLIMF? command in line 450 returns the limit test result for failed points. The test results are in the following order: sweep parameter, result (0 for fail, −1 for no test), upper limit, and lower limit.

The OUTPFAIP? command in line 470 returns the number of failed points. (When the limit test result is PASS, it returns 0 and the program goes to Passed.) Then the array Dt is printed with as many lines as the transferred data.

Lines 510 to 560 print the limit test result as follows:

```
Swp.Prmtr(Hz) : 1.1925E+7
   Result          Upper          Lower
     0              20             -40

Swp.Prmtr(Hz) : 1.2125E+7
   Result          Upper          Lower
     0              20             -40
```

What are other commands used to retrieve the test results?

Instead of reading the limit test results for failed points by using the OUTPLIMF? command, you can read out the test result using the following commands:

- At all measurement points: OUTPLIML?
- At marker position: OUTPLIMM?

Both commands return the sweep parameter, result (1 for pass, 0 for fail, −1 for no test), upper limit, and lower limit.

9

Using the List Sweep Function

This chapter describes how to use the list sweep function over HP-IB.

To Set List Sweep

```
10   !
20   ! Figure 9-1. List Sweep
30   !
40   ASSIGN @Hp4396 TO 800  ! When iBASIC is used, change "717" to "800"
50   !
60   OUTPUT @Hp4396;"SA"
70   CLEAR SCREEN
80   PRINT "Segment";TAB(9);"Center(Hz)";TAB(20);"Span(Hz)";TAB(30);"Points";
90   PRINT TAB(39);"Power(dBm)";TAB(50);"RBW(Hz)"
100  !
110  DIM Table(1:31,1:5)
120  INPUT "Enter number of segments (<=31)",Numb
130  FOR I=1 TO Numb
140    GOSUB Loadlist
150  NEXT I
160    !
170  LOOP
180    INPUT "Do you want to edit? (Y/N)",An$
190  EXIT IF An$="N" OR An$="n"
200    INPUT "Enter segment number(<=31)",I
210    IF Numb<I THEN Numb=I
220    GOSUB Loadlist
230  END LOOP
240   !
250  OUTPUT @Hp4396;"EDITLIST"
260  OUTPUT @Hp4396;"CLEL"
270  FOR K=1 TO Numb
280    OUTPUT @Hp4396;"SADD"
290    OUTPUT @Hp4396;"CENT ";Table(K,1)
300    OUTPUT @Hp4396;"SPAN ";Table(K,2)
310    OUTPUT @Hp4396;"POIN ";Table(K,3)
320    OUTPUT @Hp4396;"POWE ";Table(K,4)
330    OUTPUT @Hp4396;"BW ";Table(K,5)
340    OUTPUT @Hp4396;"SDON"
350  NEXT K
360  OUTPUT @Hp4396;"EDITDONE"
370  OUTPUT @Hp4396;"SWPT LIST"
380 !
```

Figure 9-1. Sample Program : List Sweep (1/2)

```
390  INPUT "Connect DUT, and press Enter.",Dum$
400  OUTPUT @Hp4396;"CLES"
410  OUTPUT @Hp4396;"*SRE 4;ESNB 1"
420  ON INTR 8 GOTO Sweep_end  ! \ When iBASIC is used, change "7" to "8"
430  ENABLE INTR 8;2           ! /
440    OUTPUT @Hp4396;"SING"
450 Measuring:GOTO Measuring
460 Sweep_end:   !
470 DISP "Program End"
480  STOP
490  !
500 Loadlist: !
510  INPUT "Enter center frequency(Hz)",Table(I,1)
520  INPUT "Enter frequency span(Hz)",Table(I,2)
530  INPUT "Enter number of points",Table(I,3)
540  INPUT "Enter power level(dBm)",Table(I,4)
550  INPUT "Enter resolution band width(Hz)",Table(I,5)
560  PRINT I;TAB(11);Table(I,1);TAB(20);Table(I,2);TAB(30);Table(I,3);
570  PRINT TAB(40);Table(I,4);TAB(50);Table(I,5)
580  RETURN
590  END
```

Figure 9-1. Sample Program : List Sweep (2/2)

Edit List Table

```
 70   CLEAR SCREEN
 80   PRINT "Segment";TAB(9);"Center(Hz)";TAB(20);"Span(Hz)";TAB(30);"Points";
 90   PRINT TAB(39);"Power(dBm)";TAB(50);"RBW(Hz)"
100   !
110   DIM Table(1:31,1:5)
120   INPUT "Enter number of segments (<=31)",Numb
130   FOR I=1 TO Numb
140     GOSUB Loadlist
150   NEXT I
160     !
170   LOOP
180     INPUT "Do you want to edit? (Y/N)",An$
190   EXIT IF An$="N" OR An$="n"
200     INPUT "Enter segment number(<=31)",I
210     IF Numb<I THEN Numb=I
220     GOSUB Loadlist
230   END LOOP
 :
500 Loadlist: !
510   INPUT "Enter center frequency(Hz)",Table(I,1)
520   INPUT "Enter frequency span(Hz)",Table(I,2)
530   INPUT "Enter number of points",Table(I,3)
540   INPUT "Enter power level(dBm)",Table(I,4)
550   INPUT "Enter resolution band width(Hz)",Table(I,5)
560   PRINT I;TAB(11);Table(I,1);TAB(20);Table(I,2);TAB(30);Table(I,3);
570   PRINT TAB(40);Table(I,4);TAB(50);Table(I,5)
580   RETURN
```

Lines 70 to 90 print the list table heads on the BASIC screen.

Line 110 defines the table (array `Table(1:31,1:5)`) used to hold the list values. It contains the center frequency, frequency span, number of points, power level, and resolution band width of each segment as follows:

Segment	Center Frequency	Frequency Span	Number of Points	Power Level	Resolution Band Width
1	Table(1,1)	Table(1,2)	Table(1,3)	Table(1,4)	Table(1,5)
2	Table(2,1)	Table(2,2)	Table(2,3)	Table(2,4)	Table(2,5)
⋮	⋮	⋮	⋮	⋮	⋮

Lines 130 to 150 call the subroutine `Loadlist` (line 500) to edit and print as many segments as you defined in line 120 (The analyzer can retain up to 31 segments).

```
Segment  Center(Hz)  Stop(Hz)  Points  Power(dBm)  RBW(Hz)
 1          100         20       100       0         100
 2          10000       1000     300       0         300
 3          1000000     1000     400       0         100
```

The loop, lines 170 to 230, determines if you want to edit the table and confirms that the segment is in the table.

Set List Table

```
250  OUTPUT @Hp4396;"EDITLIST"              Start to set the list table
260  OUTPUT @Hp4396;"CLEL"                  Clear existing limit table
270  FOR K=1 TO Numb                                                  Setting the list table
280    OUTPUT @Hp4396;"SADD"                Start to set a segment
290    OUTPUT @Hp4396;"CENT ";Table(K,1)    Enter center frequency
300    OUTPUT @Hp4396;"SPAN ";Table(K,2)    Enter frequency span      Setting a segment
310    OUTPUT @Hp4396;"POIN ";Table(K,3)    Enter number of points
320    OUTPUT @Hp4396;"POWE ";Table(K,4)    Enter power level
330    OUTPUT @Hp4396;"BW ";Table(K,5)      Enter resolution band width
340    OUTPUT @Hp4396;"SDON"                Complete the segment
350  NEXT K
360  OUTPUT @Hp4396;"EDITDONE"              Complete the list table
370  OUTPUT @Hp4396;"SWPT LIST"             Activate list sweep
```

In this portion of the program, the list table (edited using BASIC) is transferred to the analyzer.

What are other commands are used to set the list values?

- When setting segment frequencies, instead of setting the center/span definition by using the `CENT` *parameter* / `SPAN` *parameter* commands, you can define start/stop frequency by using:

  1. `STAR` *parameter* / `STOP` *parameter* commands
  2. `MKRSTAR` / `MKRSTOP` commands (Maker to start/stop)

- When setting the IF band width (with the analyzer in network analyzer mode), use the `BW` *parameter* command.

10

Using the Analyzer's I/O Port

This chapter describes how to use the I/O port of the analyzer with the HP-IB. For general operation of the I/O port, see the *HP 4396A Function Reference* manual.

The I/O port on the analyzer's rear panel communicates with external devices (for example, a handler on a production line).



Figure 10-1. I/O Port

The I/O port consists of the following 15 TTL compatible signals:

- 8-bit output
- 4-bit input
- 3 grounds

The signals IN 0 to IN 3 and OUT 0 to OUT 7 can be read and set by HP-IB commands.



The HP Instrument BASIC commands `READIO(15,0)` and `WRITEIO 15,0` can directly control the 8-bit I/O port without using HP-IB commands. This operation is faster than using an HP-IB command. For more information on these commands, see the *Using HP Instrument BASIC with the HP 4396A* manual.

To Synchronize External Handler with Analyzer

```
10 !
20 ! Figure 10_2. Synchronization of an External Handler
30 !              with the Analyzer
40 !
50  ASSIGN @Hp4396 TO 717 ! When iBASIC is used, change "717" to "800"
60  !
70  !
80  OUTPUT @Hp4396;"OUT8IO 8"
90  !
100 REPEAT                          !
110   OUTPUT @Hp4396;"INP8IO?"      !
120   ENTER @Hp4396;Inpio           ! Waiting Handler Response
130   A=BIT(Inpio,3)                !
140 UNTIL A=1                       !
150 !
160 !
170 END
```

Figure 10-2.
Sample Program : Synchronization of an External Handler with the Analyzer

Send Signal to the External Handler

```
80 OUTPUT @Hp4396;"OUT8IO 8"
```

The `OUT8IO` *parameter* command sets the 8-bit data value of the OUT 0-7 lines. The `OUT8IO 8` command sets the OUT 3 line to TRUE (1).



Figure 10-3. 8-Bit Data of OUT0-7



Figure 10-4. Sending Signal to an the External Handler

Read Signal from the External Handler

```
100 REPEAT                          !
110   OUTPUT @Hp4396;"INP8IO?"      !
120   ENTER @Hp4396;Inpio           ! Waiting Handler Response
130   A=BIT(Inpio,3)                !
140 UNTIL A=1                       !
```

The INP8IO? command returns the 4-bit data value of the IN 0-3 lines.

Lines 100 to 160 wait for the external handler to set signal on line IN 3 to TRUE (1).



Figure 10-5. Reading Signal from the External Handler

11

Using Application Programs

This chapter provides sample programs for your convenience when calculating the following spectrum analysis factors:

- Total Harmonic Distortion (THD)
- Adjacent Channel Power
- Occupied Bandwidth

Total Harmonic Distortion

Most transmitting devices and signal sources contain harmonics as shown in Figure 11-1.



Figure 11-1. Harmonic Distortion in a Signal

This program computes the total harmonic distortion (THD) as defined by the following equation:

$$THD = \frac{\sqrt{V_2^2 + V_3^2 + \cdots}}{V_1} \times 100[\%] \qquad (11-1)$$

Where,

V_1 Fundamental [V]
V_2 The second harmonic [V]
V_3 The third harmonic [V]
⋮

THD takes into account the power in all the harmonics. Because an infinite number of the harmonics cannot be measured, a finite number will have to suffice.

Before running the program, measure the signal and display the fundamental and harmonics to be computed on the analyzer display (in the dBm format).

```
10   !
20   ! Figure 11-2. Total Harmonic Distortion
30   !
40      Vf=1
50      ASSIGN @Hp4396 TO 717  ! When iBASIC is used, replace "717" to "800"
60      OUTPUT @Hp4396;"CLES;*SRE 4;ESNB 96"
70      ON INTR 7 GOTO Done    ! \ When iBASIC is used,
80      ENABLE INTR 7;2        ! / replace "7" to "8"
90      OUTPUT @Hp4396;"STOP?"
100     ENTER @Hp4396;Fstop
110     OUTPUT @Hp4396;"PRSMKRS"
120     OUTPUT @Hp4396;"MKR ON;SEAM PEAK"
130     OUTPUT @Hp4396;"OUTPMKR?"
140     ENTER @Hp4396;Vf,Vf2,Ff     ! Fundamental
150     Vf=SQR(10^(Vf/10)*.05)      ! Vf in V
160     PRINT "Fundamental"
170     Fr=Ff
180     I=2
190     S=0
200   LOOP
210     Fh=Ff*I
220   EXIT IF Fstop<=Fh
230     OUTPUT @Hp4396;"DMKR TRAC;MKRPRM ";Fh-Fr/2
240     OUTPUT @Hp4396;"DMKR ON"
250     OUTPUT @Hp4396;"MKRPRM ";Fr
260     OUTPUT @Hp4396;"PARS ON;SEARSTR"
270     OUTPUT @Hp4396;"SEAM PEAK;DMKR OFF"
280     OUTPUT @Hp4396;"OUTPMKR?"
290     ENTER @Hp4396;Vh,Vh2,F
300     Vh=10^(Vh/10)*.05           ! Vh^2 in V^2
310     PRINT I;" harmonic"
320     S=S+Vh
330     I=I+1
340   END LOOP
350  !
360 Done:   !
370    Thd=SQR(S)/Vf*100
380    PRINT "THD=";Thd;" %"
390    DISP "PROGRAM FINISHED"
400    END
```

Figure 11-2. Sample Program : Total Harmonic Distortion (THD)

In line 120 the marker searches for the fundamental frequency.

In lines 200 to 340 the marker searches for the harmonics on the analyzer display and integrates the squares.

Line 370 calculates the THD and line 380 prints the result.

Adjacent Channel Power Calculation

The adjacent channel power measurement examines the leakage power transmitted into an adjacent channel (that is, the channel next to the carrier channel). This program calculates the adjacent channel power leakage ratio to the power of the transmitter (Pl–Pc, Ph–Pc) in dBc.



Figure 11-3. Adjacent Channel Power

Before running the program, set up the measurement, calibrate the analyzer, and connect the signal to the input port.

```
10 !
20 ! Figure 11-4. Adjacent Channel Power Calculation
30 !
40    ASSIGN @Hp4396 TO 717    ! When iBASIC is used, change "717" to "800"
50    Fadj=25000               ! Hz
60    Fs=16000                 ! Hz
70    Rbw=100                  ! Resolution bandwidth, Hz
80    Nop=801                  ! Number of measurement points
90    Fspan=80000              ! Frequency Span, Hz
100   Avg=10                   ! Averaging factor
110   !
120   CLEAR SCREEN
130   INPUT "Enter carrier frequency(Hz).",Fcent
140   OUTPUT @Hp4396;"CENT ";Fcent
150   OUTPUT @Hp4396;"SPAN ";Fspan
160   OUTPUT @Hp4396;"BW ";Rbw
170   OUTPUT @Hp4396;"AVERFACT ";Avg
180   OUTPUT @Hp4396;"FMT NOISE;SAUNIT DBM;ATTAUTO ON;AVER ON"
190   OUTPUT @Hp4396;"HOLD;AVERREST"
200   !
210     INPUT "Connect input port and press Enter.",Dum$
220     DISP "MEASURING"
230     OUTPUT @Hp4396;"TRGS INT"
240     OUTPUT @Hp4396;"CLES"
250     OUTPUT @Hp4396;"*SRE 4;ESNB 1"
```

Figure 11-4. Sample Program : Adjacent Channel Power Calculation (1/2)

```
260     ON INTR 7 GOTO Sweep_end        ! \ When iBASIC is used,
270     ENABLE INTR 7;2                 ! / change "7" to "8"
280       OUTPUT @Hp4396;"NUMG ";Avg
290 Measuring:GOTO Measuring
300 Sweep_end:     !
310     DISP "MEASUREMENT COMPLETE"
320     DIM D(1:801)
330     OUTPUT @Hp4396;"FORM3"
340     ASSIGN @Dt TO 717;FORMAT OFF    ! When iBASIC is used,
350     OUTPUT @Hp4396;"OUTPDTRC?"      ! change "717" to "800"
360     ENTER @Dt USING "%,8A";Dum$
370     ENTER @Dt;D(*)
380     ENTER @Dt USING "%,1A";Dum$
390     !
400   Fch_l=Fcent-Fadj
410   Fch_h=Fcent+Fadj
420   Pc=FNPower(D(*),Fspan,Fcent,Fs,Nop,Fcent)
430   Pl=FNPower(D(*),Fspan,Fch_l,Fs,Nop,Fcent)
440   Ph=FNPower(D(*),Fspan,Fch_h,Fs,Nop,Fcent)
450   !
460   OUTPUT @Hp4396;"MKR ON;SMKR1 ON;SMKR2 ON"
470   OUTPUT @Hp4396;"MKRPRM ";Fcent
480   OUTPUT @Hp4396;"SMKRPRM1 ";Fch_l
490   OUTPUT @Hp4396;"SMKRPRM2 ";Fch_h
500   PRINT "Carrier (MHz):",Fcent/1.E+6
510   PRINT "Power   (dBm):",Pc
520   PRINT
530   PRINT "Adjacent Channel Freq. Lo(Hz):",Fch_l
540   PRINT "                       Hi(Hz):",Fch_h
550   PRINT
560   PRINT "Adjacent Pow. Pl-Pc(dBc):",Pl-Pc
570   PRINT "              Ph-Pc(dBc):",Ph-Pc
580   DISP "PROGRAM FINISHED"
590   END
600   !
610   DEF FNPower(D(*),Fspan,Fch,Fs,Nop,Fcent)
620     Fdelta=Fspan/(Nop-1)
630     Ich=(Fch-Fcent)/Fdelta+401
640     I1=Ich-Fs/2/Fdelta
650     I2=Ich+Fs/2/Fdelta
660     IF I1<1 OR I2>Nop THEN
670       P=0
680       RETURN P
690     END IF
700     S=0
710     FOR I=I1 TO I2
720       S=S+10^(D(I)/10)          !  S in mW
730     NEXT I
740     P=S*(Fspan/(Nop-1))
750     P=10*LGT(P)                 ! P in dBm
760     RETURN P
770   FNEND
```

Figure 11-4. Sample Program : Adjacent Channel Power Calculation (2/2)

Lines 50 to 100 set the measurement coefficient, Fadj, Fs, frequency span, resolution bandwidth, number of measurement point, and averaging factor to typical values.

In lines 610 to 760, the subprogram `FNPower` performs summation of the powers by measurement points, in the area of center frequency, `Fch`, and the frequency span, `Fs`. The total power is calculated by the following equation:

$$PWR = 10\log_{10}\sum_{x=I1}^{x=I2}\frac{D(x)Fspan}{Nop}\ [\mathrm{mW}] \qquad (11-2)$$

Where,

$D(x)$	Power density spectrum [dBm/Hz]
$Fspan$	Measurement frequency span [Hz]
Nop	Number of measurement points
I1, I2	The measurement point of left and right edge of a channel bandwidth(Fs)

Occupied Power Bandwidth Calculation

This program calculates the occupied bandwidth of the carrier signal. It first computes the combined power of all signal responses contained in the trace. It then calculates the point for which 0.5 % of the total power lies to the right of the right maker and to the left of the left marker.



Figure 11-5. 99 % Occupied Power Bandwidth

Before running the program, set up the measurement, calibrate the analyzer, and connect the signal to the input port.

```
10 !
20 ! Figure 11-6. Occupied Power Bandwidth Calculation
30 !
40    ASSIGN @Hp4396 TO 717 ! When iBASIC is used, change "717" to "800"
50    Rbw=100          ! Resolution bandwidth, Hz
60    Nop=801          ! Number of measurement points
70    Fspan=80000.     ! Frequency Span, Hz
80    Avg=10           ! Averaging factor
90    !
100   INPUT "Enter carrier frequency(Hz).",Fcent
110   OUTPUT @Hp4396;"SPAN";Fspan
120   OUTPUT @Hp4396;"CENT";Fcent
130   OUTPUT @Hp4396;"BW";Rbw
140   OUTPUT @Hp4396;"FMT NOISE;DET POS"
150   OUTPUT @Hp4396;"SAUNIT DBM;ATTAUTO ON;AVER ON"
160   OUTPUT @Hp4396;"AVERFACT";Avg
170   !
180   INPUT "Connect input port and press Enter.",Dum$
190   OUTPUT @Hp4396;"HOLD"
200   OUTPUT @Hp4396;"CLES"
210   OUTPUT @Hp4396;"*SRE 4;ESNB 1"
220   ON INTR 7 GOTO Sweep_end     ! \ When iBASIC is used,
230   ENABLE INTR 7;2              ! / change "7" to "8"
240     OUTPUT @Hp4396;"TRGS INT"
250     OUTPUT @Hp4396;"NUMG";Avg
260 Measuring:GOTO Measuring
270 Sweep_end:  !  Get Data
280   DIM D(1:801)
290   DIM P(1:801)
300   OUTPUT @Hp4396;"FORM3"
310   ASSIGN @Dt TO 717;FORMAT OFF ! When iBASIC is used,
320   OUTPUT @Hp4396;"OUTPDTRC?"   ! change "717" to "800"
330   ENTER @Dt USING "%,8A";Dum$
340   ENTER @Dt;D(*)
350   ENTER @Dt USING "%,1A";Dum$
360   !
370   Power(D(*),P(*),Rbw,Nop,Fspan)
380   !
390   FOR I=1 TO Nop
400     A=P(I)/P(Nop)
410     IF A>.005 THEN Lower
420   NEXT I
430 Lower:I1=I
440   FOR I=Nop TO 1 STEP -1
450     A=P(I)/P(Nop)
460     IF A<.995 THEN Upper
470   NEXT I
480 Upper:I2=I
```

Figure 11-6. Sample Program : Occupied Power Bandwidth Calculation (1/2)

```
490     OUTPUT @Hp4396;"MKR ON"
500     OUTPUT @Hp4396;"MKRP ";I1
510     OUTPUT @Hp4396;"DMKR ON"
520     OUTPUT @Hp4396;"MKRP ";I2
530     OUTPUT @Hp4396;"OUTPMKR?"
540     ENTER @Hp4396;Val,Val2,Flh
550     PRINT "Occupied bandwidth :";
560     PRINT Flh;" Hz"
570   DISP "PROGRAM FINISHED"
580   END
590   !
600   SUB Power(D(*),P(*),Rbw,Nop,Fspan)
610     S=0
620     FOR I=1 TO Nop
630       S=S+10^(D(I)/10)      !  S in mW
640       P(I)=S
650     NEXT I
660   SUBEND
```

Figure 11-6. Sample Program : Occupied Power Bandwidth Calculation (2/2)

Lines 40 to 80 set the measurement coefficient, frequency span, resolution bandwidth, number of measurement points, and averaging factor to typical values.

Lines 390 to 430 search from the left for the point where the power is 0.5 % compared to the total power. Lines 440 to 480 do the same search from the right.

Lines 490 to 540 display the marker and Δ marker on the 0.5 % power point and read out the spacing of the markers.

Lines 600 to 660 (subprogram `Power`) perform a summation of the power at the measurement points. This summation is done in the area of the center frequency (`Fch`) and the frequency span (`Fs`). The same equation (11-2) is used in the "Adjacent Channel Power Calculation" example.

12

If You Have a Problem

This chapter provides the information you need to correct the listed problems.

If There Is No Response From an Instrument on the HP-IB Bus

- ☐ Check all HP-IB addresses and cable connections.

  Most HP-IB problems are caused by an incorrect address or a bad or loose HP-IB cable.

If an Error Message is Displayed

- ☐ Check the error message on the analyzer's display.



- ■ If "`HP-IB error occurred`" is displayed:

  1. Get the error number and description using the `OUTPERRO?` command. (For information on how to use this command, see the "To Report Command Error Occurrence" in Chapter 3.)
  2. See "Messages" in the *HP-IB Command Reference* manual.

- ■ If any other message is displayed:

  See "Messages" in the *HP-IB Command Reference* manual.

If the Disk Cannot Be Read

□ Check the disk.

1. Put the disk into the disk drive and type as follows.

```
CAT
```

2. Press Return.

   If error message is displayed, the disk is corrupted or the disk format does not match. Use another disk.



If you are using the external controller,

HP BASIC uses the LIF format, but doesn't use DOS format. HP Instrument BASIC uses both the LIF and the DOS format. Try again on using HP Instrument BASIC.

□ Check the mass storage.

1. Put the disk into the disk drive and type as follows:

```
SYSTEM$("MSI")
```

2. Press Return.

```
:CS80, 700, 0        ←—mass storage volume specifier
```

3. If the mass storage volume does not match your disk drive, use the `MSI` statement to set it to match.



If you are using HP Instrument BASIC:

To use the built-in disk drive, mass storage volume specifier must be `:,4`.

□ Check the file type.

1. Put the disk into the disk drive and type as follows:

```
CAT
```

2. Press Return.

```
CAT
      :
FILE NAME  PRO TYPE REC/FILE BYTE/REC   ADDRESS    DATE     TIME

FIG1_3        ASCII       6       256        34   22-Jun-92 11:00
FIG2_2        PROG        6       256        34   22-Jun-92 11:00
FIG2_3        ASCII       6       256        34   22-Jun-92 11:00
      :
```

If you are using HP Instrument BASIC

Only an ASCII type program file can be saved and read.

Use the SAVE/GET commands to save and read ASCII files.



If you are using HP BASIC

- To read ASCII type program, use GET command.

  The ASCII type program file can be saved and read using SAVE/GET commands.

- To read a PROG type program, use the READ command.

  The PROG type program file can be saved and read using the STORE/LOAD commands.

If the HP-IB Command Does Not Work

□ Check the preceding HP-IB command.

An HP-IB command that requires execution time (such as changing format or calculating the calibration coefficients) can cause next HP-IB command to fail.

If you are using such commands, insert the following command lines:

```
OUTPUT @Hp4396;"*OPC?"
ENTER @Hp4396;Dum
```

For details, see "To Wait For the Preceding Operation to Complete" in Chapter 3.

A

Manual Changes

Introduction

This appendix contains the information required to adapt this manual to earlier versions or configurations of the analyzer than the current printing date of this manual. The information in this manual applies directly to the HP 4396A Network/Spectrum Analyzer serial number prefix listed on the title page of this manual.

Manual Changes

To adapt this manual to your HP 4396A, see Table A-1 and Table A-2, and make all the manual changes listed opposite your instrument's serial number and firmware version.

Instruments manufactured after the printing of this manual may be different from those documented in this manual. Later instrument versions will be documented in a manual changes supplement that will accompany the manual shipped with that instrument. If your instrument's serial number is not listed on the title page of this manual or in Table A-1, it may be documented in a *yellow MANUAL CHANGES* supplement.

In additions to change information, the supplement may contain information for correcting errors (Errata) in the manual. To keep this manual as current and accurate as possible, Hewlett-Packard recommends that you periodically request the latest *MANUAL CHANGES* supplement.

For information concerning serial number prefixes not listed on the title page or in the *MANUAL CHANGE* supplement, contact the nearest Hewlett-Packard office.

Turn on the line switch or execute the `*IDN?` command by HP-IB to confirm the firmware version. See the *HP-IB Command Reference* manual for information on the `*IDN?` command.

Table A-1. Manual Changes by Serial Number

Serial Prefix or Number	Make Manual Changes

Table A-2. Manual Changes by Firmware Version

Version	Make Manual Changes

Serial Number

Hewlett-Packard uses a two-part, nine-character serial number that is stamped on the serial number plate (see Figure A-1) attached to the rear panel. The first four digits and the letter are the serial prefix and the last five digits are the suffix.



Figure A-1. Serial Number Plate

Index

REGIONAL SALES AND SUPPORT OFFICES

For information relating to Sales or Support of Hewlett-Packard products first contact your local Hewlett-Packard office listed in the white pages of your telephone directory. If none is listed locally, contact one of the addresses listed below to obtain the address or phone number of the Hewlett-Packard Sales or Support office nearest you.

ASIA

Hewlett-Packard Asia Ltd.
47/F, 26 Harbour Road,
Wanchai, **HONG KONG**
G.P.O. Box 863, Hong Kong
Tel: (852)5-8330833
Telex: 76793 HPA HX
Cable HPASIAL TD

AUSTRALASIA

Hewlett-packard Australia Ltd.
31-41 Joseph Street
BLACKBURN, Victoria 3130
Australia
Tel: (61)895-2895
Telex: 31-024
Cable: HEWPARD Melbourne

CANADA

Hewlett-Packard (Canada) Ltd.
6877 Goreway Drive
MISSISSAUGA, Ontario L4V 1M8
Tel: (416)678-9430
Telex: 069-8644

JAPAN

Yokogawa-Hewlett-Packard Ltd.
29-21 Takaido-Higashi, 3 Chome
Suginami-ku **TOKYO** 168
Tel: 03(331)6111
Telex: 232-2024 YHPTOK

MEDITERRANEAN AND MIDDLE EAST

Hewlett-Packard S.A.
Mediterranean and Middle East
Operations
Atrina Centre
32 Kifissias Avenue
Paradissos-Amarousion, **ATHENS**
Greece
Tel: (30)682 88 11
Telex: 21-6588 HPAT GR
Cable: HEWPACKSA Athens

May, 1990

BENELUX & SCANDINAVIA

Hewlett-Packard S.A.
Uilenstede 475
P.O. Box 999
NL-1183 AG **AMSTELVEEN**
The Netherlands
Tel: (31)20/43 77 71
Telex: 18 919 hpner nl

SOUTH & EAST EUROPE, AFRICA

Hewlett-Packard S.A.
7 rue du Bois-du-Lan
CH-1217 **MEYRIN** 2, Geneva
Switzerland
Tel: (41)22/83 12 12
Telex: 27835 hmea
Cable: HEWPACKSA Geneva

FRANCE

Hewlett-Packard France
Parc d'activites du Bois Briard
2, avenue du Lac
91040 **EVRY** Cedex
Tel: 1 6/077 83 83
Telex: 6923 15F

GERMAN FEDERAL REPUBLIC

Hewlett-Packard GmbH
Hewlett-Packard-Strasse
Postfach 1641
D-6380 **BAD HOMBURG**
West Germany
Tel: 06172/400-0
Telex: 410 844 hpbhg

ITALY

Hewlett-Packard Italiana S.p.A.
Via G. Di Vittorio 9
1-20063 **CERNUSCO SUL NAVIGLIO**
(Milano)
Tel: 02/92 36-91
Telex: 334632

UNITED KINGDOM

Hewlett-Packard Ltd.
King Street Lane
Winnersh, **WOKINGHAM**
Berkshire RG11 5AR
Tel: 734/78/ 47 74
Telex: 847178

EASTERN USA

Hewlett-Packard Co.
4 Choke Cherry Road
ROCKVILLE, MD 20850
TEL: (301)670-4300

MIDWESTERN USA

Hewlett-Packard Co.
5201 Tollview Drive
ROLLING MEADOWS, IL 6008
Tel: (312)255-9800

SOUTHERN USA

Hewlett-Packard Co.
2000 South Park Place
P.O. Box 105005
ATLANTA, GA 30348
Tel: (404)955-1500

WESTERN USA

Hewlett-Packard Co.
5161 Lankershim Blvd.
P.O. Box 3919
NO. HOLLYWOOD, CA 91609
Tel: (818)506-3700

OTHER INTERNATIONAL AREAS

Hewlett-Packard Co.
Intercontinental Headquarters
3495 Deer Creek Road
PALO ALTO, CA 94304
Tel: (415)857-1501
Telex: 034-8300
Cable: HEWPACK